

פרל Perl

מדריך למתכנת

מהדורה שנייה

נכתב ע"י שמואל פומברג
בתאריך 20.5.2008
כל הזכויות שמורות 2003-2008
(ראה תנאי רישיון בפרק 1.5)

<http://code.semuel.co.il/perlhebtut/>

תוכן עניינים

4	מילות פתיחה	1
4	למה פרל?	1.1
4	למי המאמר מיועד?	1.2
4	איך לומדים את השפה הזאת?	1.3
5	מה אני צריך כדי לתכנת בפרל?	1.4
5	תנאי רישיון	1.5
5	אז בואו נתחיל כבר	1.6
7	סוגי משתנים והגדרתם	2
7	כללי	2.1
7	הגדרה ספונטנית	2.2
7	הגדרת משתנה מקומי	2.3
8	משתנים סקלריים	2.4
9	מערכים	2.5
10	טבלת האש	2.6
11	הקשרים ומשתנים	2.7
11	תוכנית דוגמא	2.8
12	מבני בקרה	3
12	ביטויים בוליאניים	3.1
13	פקודת if וניגזרותיה	3.2
13	פקודת while	3.3
14	לולאת for	3.4
15	פעולות על תנאי	3.5
15	תוכנית דוגמא	3.6
17	טיפול במחרוזות	4
17	אופרטורים של מחרוזות	4.1
17	מחרוזות ומשתנים	4.2
18	ביטויים רגולריים	4.3
20	פקודת התאמת מחרוזות	4.4
21	פקודת החלפה	4.5
21	חלק ממחרוזות	4.6
21	מציאת מיקום בתוך מחרוזת	4.7
22	חיבור מחרוזות	4.8
22	חיתוך שורה חדשה	4.9
22	סיכום מחרוזות	4.10
23	תוכנית דוגמא	4.11
24	פונקציות והוראות	5
24	משתני ברירת מחדל	5.1
24	כתיבת פונקציות	5.2
26	הוראות לקומפילר	5.3
27	תוכנית דוגמא	5.4
29	טיפול בקבצים	6
29	הכוונת קלט/פלט	6.1
29	פתיחת קובץ	6.2
30	אופרטור <>	6.3
30	כתיבה לקובץ	6.4
31	סגירת קובץ	6.5
31	תוכנית דוגמא	6.6
32	מודולים	7

32 כללי	7.1
32 שימוש במודולים	7.2
32 מודולים כאובייקטים	7.3
33 מודולים פופולריים	7.4
33 תוכניות דוגמא	7.5
35 אינטרנט	8
35 הקדמה	8.1
35 פרוטוקול HTTP על קצה המזלג	8.2
36 המודול CGI	8.3
36 טפלייטים	8.4
37 דפים משולבי קוד	8.5
38 ומה עוד	8.6
39 עברית בפרל	9
39 הקדמה	9.1
39 מושגי יסוד	9.2
39 יוניקוד	9.3
40 חזרה לפרל	9.4
41 זרימת טקסט	9.5
41 זהירות – הגדרות	9.6
41 ביטויים רגולריים	9.7
42 תוכנית דוגמא	9.8
43 תרבות וסביבה	10
43 מלחמות דת	10.1
43 גישה	10.2
43 שיתוף ורשיונות	10.3
44 גרסאות של פרל	10.4
45 מידע נוסף ועזרה	11
45 למה צריך עוד מידע	11.1
45 ספרים	11.2
45 רשימות תפוצה	11.3
45 אתרים	11.4
45 ערוצי צ'אט באינטרנט	11.5
46 התייעוד	11.6
46 פרקי קריאה	11.7
48 סיכום	12
48 תודות	12.1
48 ביבליוגרפיה	12.2
48 על מסמך זה	12.3
49 נספח א' – מלכודות למתכנת המתחיל	
50 נספח ב' - דברים נוספים שכדאי להכיר	
50 כללי	
50 גרשיים	
52 בלוקים	
53 lvalue	
54 מרחבי שמות	
55 נספח ג' - רשימת פונקציות מובנות	

1. מילות פתיחה

1.1. למה פרל?

פרל (Perl) היא שפה המיועדת במקור לעיבוד טקסטים. נשמע מעט? לא אם זוכרים שכל האינטרנט מורכבת, בעצם, מטקסטים. הוסף לכך פקודות רבות המשותפות עם שפת C, המון מודולי הרחבה שמקלים את החיים וכו', ויש לך נוסחה מנצחת. דבר נוסף זה זמן הפיתוח. לכתוב תוכנית בפרל (נורמאלית, לא כזו שרק כותבת "שלום") לוקח במקרים רבים עשירית מהזמן שייקח לכתוב תוכנית C מקבילה. פרל רצה, למעשה, תחת כל סביבה, תחת כל מערכת הפעלה. אם אתה צריך להריץ על יוניקס, לינוקס, חלונות, או מקינטוש, תוכל לעשות זאת.

1.2. למי המאמר מיועד?

המאמר מיועד למתחילים, בעלי רקע בסיסי בתכנות, בשפת כלשהי. (כגון שפת C, או Pascal) לא נסביר כאן מהו משתנה, מהו סוג משתנה, מהו מערך, טבלת-האש או פונקציה, מה ההבדל בין קיר לעיר, או כל דבר אחר. אני מניח שאתה כבר עברת על דברים כאלה.

1.3. איך לומדים את השפה הזאת?

קודם כל, יש את המדריך שאתה קורא כרגע. זו כבר התחלה טובה. יש גם את התיעוד של פרל עצמה, המגיע ביחד עם השפה. ראה פרק 11.6 להסבר. בגדול, יש שני מקורות מהם אפשר ללמוד פרל: אינטרנט וספרים. שמות של ספרים למתחילים אתה יכול למצוא בפרק 11, תחת הכותרת "ספרים". יש כמובן ספרים למתקדמים, וספרים על נושאים ספציפיים בשפה, אבל בוא נשאיר את זה לכשתגיע לרמה הזאת. (בפרק 11 יש עוד כמה הפניות לעזרה, אבל בוא נשאיר את הפרק לזמנו.) ובאינטנט, יש הרבה מקורות מהם אפשר ללמוד. האתרים המרכזיים הם www.perl.com, www.cpan.com, ועוד. ואם אתה מתעקש על חינוך פורמלי, יש קורסים. באתר <http://code.semuel.co.il/perl> יש הפניות לקורסים בארץ. ולבסוף, הדבר הכי יעיל ללמידה זה סבלנות ונכונות לניסויים. אם אתה לא מבין את ההסבר על איזו פונקציה, הדרך הכי קלה היא פשוט לנסות אותה. ואז לשנות את הפרמטרים, ולנסות שוב. אתה תופתע לגלות כמה אנשים, אפילו מנוסים, מנסים דברים. כי התשובה המהירה ביותר לשאלה, "איך פרל יתנהג אם אני אעשה את זה?", היא פשוט לנסות את זה. אם אחרי כמה ניסיונות כאלה עדיין לא הבנת, תשאל אנשים. למשל ברשימת התפוצה (www.perl.org.il).

1.4. מה אני צריך כדי לתכנת בפרל?

השפה נכתבת בכל עורך טקסט פשוט. אם זה פנקס-הרשימות של חלונות, או pico של יוניקס, לא משנה. אפשר גם להשתמש בעורך של שפות אחרות, אם הוא לא מפריע. העיקר שלא יתווספו לקובץ שמתקבל תווי בקרה נוספים. מה שאתה רואה, זה מה אנחנו רוצים לקבל בקובץ. בשביל להריץ תוכנית פרל, אתה צריך את התוכנה. בהתקנות יוניקס/לינוקס התוכנה בד"כ כבר מותקנת כברירת מחדל. עבור חלונות, ניתן להוריד מהאינטרנט בחינם את הגירסה של פרל עבור חלונות המופצת ע"י חברת ActiveState בכתובת:

<http://www.activestate.com/Products/ActivePerl/>

עבור כל סביבה אחרת, חפש באתר:

<http://www.cpan.org/ports>

מה שבטוח, זה לא עולה כסף.

1.5. תנאי רישיון

המדריך מופץ תחת GNU FDL, (גירסא 1.2) שזה אומר שמותר לך להדפיס, להפיץ ולשכפל את המדריך, תחת התנאים הבאים:

א. המדריך מופץ בשלמותו, ללא שינויים.

ב. מופיע על הכריכה או במקום בולט אחר קישור לדף האינטרנט המקורי של המדריך.

(שזה <http://code.semuel.co.il/perlhebtut/>)

אם ברצונך לשים את המדריך על אתר, נא צור קשר איתי. תודה. (owner@semuel.co.il)

1.6. אז בואו נתחיל כבר

פתח בעורך החביב עליך קובץ חדש הנקרא, נגיד, first.pl. שים לב לסיומת - לא בכל מקום היא חובה (למשל, יוניקס לא מתחשב בסיומות), אבל תמיד טוב לסמן את הקבצים שלך. כתוב את התוכנית הבאה:

```
#!/usr/bin/perl -w
use strict;
print "Hello world!\n";
```

השורה הראשונה אומרת למחשב שעליו לקרוא לתוכנית פרל (במיקום המצויין בשורה) לתרגם את הקובץ. המתג (switch, ניקרא גם דגל בעיברית) "-w" מורה לפרל לפעול במצב של אזהרות - הקומפילר יזהיר מפני דברים שהוא לא היה אומר אם לא היה את המתג הזה.

השורה השניה היא שורת "use". use בפרל מקביל ל-include, pragma ב-C ועוד כל מיני. דרך פקודות use אפשר להורות לפרל להשתמש במודולים שונים, להרשות שימוש רק בגירסאות מתקדמות יותר של פרל ממספר גירסה מסויים, או להורות לפרל להכנס למצבים מסוימים. במקרה שלנו, ה-strict אוסף היצמדות לחוקים נוקשים של כתיבת התוכנית, למשל מצב זה אוסר הגדרה ספונטנית של משתנים. נדבר על כך עוד בהמשך.

בתור מתחילים (וגם בעתיד בתור מתקדמים) מומלץ *מאוד* להשתמש במתג ה-"-w" וב-"use strict" שכן הם ביחד יתפסו הרבה טעויות שאתם תעשו ויעזרו לכם בדיבוג התוכניות שלכם. השורה השלישית אומרת לכתוב "Hello World", כאשר ה-"\n" בסוף הוא התו המסמן מעבר שורה. פשוט.

כדי להריץ ממש את התוכנית הראשונה שלך, כתוב בשורת הפקודה של הטרמינל שאתה עובד איתו:

```
chmod a+x first.pl
./first.pl
```

לא נעבור כאן על הסבר מפורט של פקודות אלו שכן זהו שיעור פרל ולא יוניקס.
אם הנך עובד תחת חלונות, אזי בהנחה שהנתיב אל פרל הוא "C:\perl\bin\perl", אזי ההרצה מחלון
ה-DOS של חלונות (ניקרא גם חלון הפקודות או ה-Command Prompt) תהיה ע"י הפקודה:
c:\perl\bin\perl -w first.pl

2. סוגי משתנים והגדרתם

2.1. כללי

בפרל קיימים ארבע סוגי משתנים: סקלר, מערך, טבלת האש וקובץ. הסוגים נבדלים אחד מהשני באמצעות התו שמקדים אותם. לדוגמא: \$m הוא סקלר. (אם עכשיו אתה ממלמל "בייסיק", אני לא אתווכח) קיימים שלושה דרכים להגדיר משתנה: מקומית, גלובלית (עליה נדבר בפרק אחר) וספונטנית.

2.2. הגדרה ספונטנית

הגדרה ספונטנית היא הגדרת משתנה תוך כדי קוד. אם באמצע הקוד תכתוב את השורה:
\$m=5;

אזי פרל תגדיר משתנה חדש, סקלר בשם \$m ותשים בתוכו את המספר 5. היתרון היחיד שאפשר לחשוב עליו זה שהכתיבה חופשית, ולא צריך להגדיר כל משתנה, אלא אפשר להמציא משתנים תוך כדי כתיבה. החסרונות הם שזה מכוער, זה לא קריא, אתה עלול לטעות בכתיב, (שאיז פרל ימציא משתנה חדש, ולך תבין מה קרה לערך של המשתנה הקיים) קשה לדעת איפה המשתנה ייהרס ואיפה הוא ימשיך לחיות, ועוד. בקיצור, אל תעשה את זה. יותר טוב, תמנע את זה מעצמך. תכתוב בתחילת הקובץ:
use strict;
ואז פרל יהיה מנוע מלהמציא משתנים חדשים. כשתנסה להמציא משתנה, הוא ייעצר (בשלב הקומפילציה) ויגיד שהמשתנה לא קיים.

2.3. הגדרת משתנה מקומי

זה לא טוב להמציא משתנה באמצע קוד. אז איך מצהירים על משתנה? פשוט:
my \$m;
זוהי הצהרה רגילה של סקלר. כמו כל שורת int i ב-C. והיא גם מצייתת לכללים זהים. אם תשים את בתוך בלוק, אזי הוא יושמד בסוף הבלוק. עוד דוגמאות:
דוגמא להגדרה עם הצבה:

my \$m=5;
דוגמא להגדרה של כמה משתנים בבת אחת:
my (\$m, \$s, \$gfsd);
דוגמא בה מוגדרים שני משתנים, באחד מוצב הערך 5, בשני מוצבת מחרוזת.
my (\$m, \$s)=(5,"Moshe");
שים לב שכל שורה מסתיימת בנקודה-פסיק. (;) נקודה-פסיק מסמנת סוף שורה/פקודה בפרל, בדיוק כמו ב-C.

2.4. משתנים סקלריים

בשפות תוכנה אחרות, יש כל מיני סוגי משתנים. שלמים, נקודה-צפה, בוליאניים, תווים ומחרוזות. בפרל קיים רק את הסקלר. (Scalar) הסקלר מכיל את כל המספרים, ואת כל המחרוזות, לא משנה באיזה אורך או רמת דיוק. הסקלר מאופיין ע"י סימן הדולר (\$) בתחילתו. דוגמאות:

```
$m=5;
$m2="Have a nice day";
$zfdg='a';
$fnlpi=3.14157265;
$stam="7";
$van=$stam+$m;
```

שימו לב לאחרון: אני מחבר את המחרוזת "7" עם המספר 5. מה שקורה זה תרגום אוטומטי. אם מנסים להפעיל פעולה אריתמטית על מחרוזת, פרל בודק מה יש בפנים. אם המחרוזת היא בעצם מספר, אזי היא מתורגמת למספר ועליו מתבצעת הפעולה. אם המחרוזת מכילה גם מספר וגם תוים אחריו, פרל בשקט (אלא אם כן השתמשת ב-"w") יוריד את התוים וישאיר את המספר. אם המחרוזת מכילה תוים בלבד, היא תתורגם לערך 0 (אפס). לייצוג ערכים בוליאניים, בפרל 0 זה שקר, כל דבר אחר זה אמת. מחרוזות - מחרוזת ריקה זה שקר, כל מחרוזת מלאה אחרת זה אמת.

הערה - המחרוזת "0" היא שקר, כיוון שהיא מתורגמת למספר 0, שהוא שקר. אפשר לעקוף את זה, ולהוציא את המחרוזת "0". שים לב לרווח לפני האפס. הרווח מונע מהמחרוזת להיתרגם אוטומטית לאפס, ובבדיקה בוליאנית המחרוזת תהיה שווה לאמת. אם תפעילו על המחרוזת פעולה אריתמטית, המחרוזת תיתרגם לאפס. ככה מוציאים אפס-אמת. דוגמא לחתיכת קוד חסרת תועלת:

```
#!/usr/bin/perl -w
use strict;
my ($a,$b,$c,$d)=(1,2,3);
$d = $a+$b*$c/$a;
print "And the number is: $d !\n";
```

כמו שאתם כבר מנחשים, כל הפעולות האריתמטיות הרגילות פועלות כרגיל.

הערה - בקטע הקוד הצבנו ערך לתוך \$d אחרי ההצהרה. מה היה בתוך \$d אחרי ההצהרה ולפני ההצבה?

התשובה היא, שאז היה בתוך \$d את הערך undef, או בעברית, לא-מוגדר. זו הדרך של פרל להגן על המתכנת מפני השתמשות במשתנה שלא אותחל. אם תנסה לעשות פעולה כלשהי על undef, תקבל הודעת שגיאה. מצד שני, אפשר להציב undef לתוך סקלר, ואפשר לבדוק האם הערך שבתוך סקלר מוגדר, בעזרת הפונקציה defined. חתיכת קוד:

```
#!/usr/bin/perl -w
use strict;
my $m;
if (defined($m)) {
    print $m;
} else {
    print "Undefined!";
}
```

לבסוף, שים לב שגם undef בבדיקה בוליאנית הוא שקר.

אופרטורים אריתמטיים :

תוצאה	שם	אופרטור
$2 + 3 = 5$	חיבור	$\$a + \b
$4 - 3 = 1$	חיסור	$\$a - \b
$2 * 3 = 6$	כפל	$\$a * \b
$17 \% 3 = 2$	מודולו (שארית)	$\$a \% \b
$8 / 4 = 2$	חילוק	$\$a / \b
$2 ** 3 = 8$	חזקה	$\$a ** \b
$\$a = \$a - 1$	הפחתה-עצמית	$\$a--$, $--\$a$
$\$a = \$a + 1$	ייסוף-עצמי	$\$a++$, $++\$a$

2.5 מערכים

אוקי, יש לנו משתנים סקלרים. עכשיו צריך לאגד אותם למערכים. המערך מסומן בשטרודל בתחילתו, (@) ומכיל כל כמות של משתנים סקלרים, ורק משתנים סקלרים. אם תנסה להכניס מערך לתוך מערך, תקבל מערך אחד ארוך. (אפשר להכניס דברים אחרים לתוך מערך, אבל באמצעות מצביעים - מה ששייך לספר אחר)

```
my @array=(4,5,7,3,3,71,"Samba!",undef,3.14,$m,@old_array,83);
```

מה שיצא מהדוגמא הזאת, זה מערך אחד גדול, המכיל את מספרים, מחרוזות, ערך שבתוך \$m כל הערכים שבתוך @old_array ואפילו undef אחד.

מערכים בפרל הם בעלי גודל משתנה - כלומר, אפשר להוסיף לסופם, לתחילתם או לכל מקום באמצע איברים נוספים, ואפשר גם להוציא איברים.

```
my @arr = (1,2,3);
my $avar = 5;
push (@arr, $avar++); # Add to the end. @arr=(1,2,3,5);
unshift (@arr, $avar++); # Add to the beginning. @arr=(6,1,2,3,5);
$avar = pop (@arr); # Remove from the end. @arr=(6,1,2,3), $avar=5.
$avar = shift(@arr); # Remove from the beginning. @arr=(1,2,3), $avar=6
```

שים לב שהסימן # מציין הערה, הנמשכת עד סוף השורה. (שלא כמו ב-C, פה הערות לא יכולות להפרס על כמה שורות, אלא כל הערה היא שורה בודדת)

גישה לערכים בתוך מערך : אפשר לגשת לאיברי המערך בבודדים, או לקבוצה שלמה, בצורה של תת-מערך.

```
$arr[3]=7;
@arr[4,5]=(32,34);
```

שים לב שבדוגמא הראשונה, \$arr[3] מתחיל עם סימן דולר, ולא שטרודל. זה כיוון שבדוגמא זו אנו מתייחסים לאיבר ספציפי בתוך המערך, שהוא בפני עצמו סקלר, ולכן הוא מתחיל בסימן דולר. בדוגמא השנייה, אני מתחיל בסימן השטרודל, כיוון שאני מתייחס לחתיכה מהמערך, שהיא מערך קטן בפני עצמה, ולכן מגיע לה שטרודל.

אני רוצה להזכיר כאן דוגמא מהסעיף הקודם :

```
my ($a,$b,$c,$d)=(1,2,3);
```

אם תעיף מבט מקרוב, תראה שבעצם מוגדרים פה לא איברים בודדים, אלא מערך, בעל ארבעה איברים. (בעצם לא מערך, אלא רשימה, אבל זה סיפור אחר) ולתוך המערך הזה מוצבים שלושה איברים. מה שיוצא זה ששלושת המשתנים הראשונים מקבלים ערך, והאחרון נשאר undef. נראה את הדוגמא הזאת שוב, עם הבדל קטן :

```
my ($a,@b,$c,$d)=(1,2,3);
```

הפעם @b הוא מערך, ולא סקלר. מה שיקרה הפעם זה: \$a=1, @b=(2,3) ו-\$c=\$d=undef. זה קורה בגלל ש-@b הוא מערך, ולכן ההצבה מכניסה אליו את כל האיברים שנותרו, ולא משאירה ל-\$c, \$d איברים. דוגמא אחרת:

```
$a=@arr;
```

כן, סקלר שווה למערך. מה שיוצא, זה שהסקלר מקבל את גודל המערך. אם היו במערך חמישה איברים \$a יקבל את הערך 5. יש עוד שתי דרכים לקבל את מספר האיברים במערך:

```
$size = scalar(@arr);  
$size = $#arr +1;
```

בשתי השורות \$size יקבל את אותו הערך כמו בדוגמא הקודמת. שים לב לתוספת 1 בשורה השנייה. זה כי "\$#arr" נותן את מספר האיבר האחרון המערך. ומכיוון שהמיספור מתחיל מאפס, אז צריך להוסיף 1 כדי להגיע למספר האיברים במערך. מה יקרה אם המערך ריק? תנסה לבד. הערה – בעמוד הקודם ציינו שהסימן '#' מסמל תחילת הערה. בכל זאת זה משמש גם לגודל מערך. תחיה עם זה. ככה שולפים את האיבר האחרון במערך:

```
$a = $arr[$#arr];
```

לבסוף, פקודת splice לטיפול במערך. פקודה קצת מסובכת אך שימושית לעיתים, המסוגלת להחליף את push, pop וכל שאר הפקודות של הכנסת והוצאת האיברים במערך. הנה דוגמאות לשימוש בפונקציה:

```
my @arr1 = (1,2,3,4,5); my @arr2=(6,7,8); my @arr3;  
@arr3=splice(@arr1,2,2,@arr2);
```

הפקודה הזאת מורידה את האיברים 3,4 מתוך @arr1, ומכניסה במקומם את איברי @arr2. מקבל את האיברים שהוסרו, ומה שיוצא זה:

```
@arr1 = (1,2,6,7,8,5), @arr2=(6,7,8), @arr3=(3,4)
```

הכנסת והוצאת איבר בעזרת splice:

```
splice(@arr1,3,0,2.5); # @arr1=(1,2,6,2.5,7,8,5)  
my $val=splice(@arr1,5,1); # @arr1=(1,2,6,2.5,7,5), $val=8
```

רק נסביר, ש-0 מסמל לפני האיבר הראשון, 5 אחרי האיבר החמישי, -1 זה לפני האיבר האחרון.

2.6. טבלת האש

או בלעז, Hash-Table. זהו מבנה נתונים דומה למערך, רק שבמקום להשתמש במספר כדי לציין מקום של איבר במערך, לכל איבר יש מפתח, שהוא יכול להיות מחרוזת או מספר (אבל לא מצביע) שהוא המציינן של האיבר. (שהוא סקלר כלשהו) לסוג נתונים זה יש את הסימן % בתחילת שם המשתנה.

דוגמא להגדרה של האש:

```
my %h=( 1 => "elf", "Zambura" => 5, 5=>7, "another"=>7);
```

זוהי טבלת האש עם ארבעה איברים, עם המפתחות ("elf",5,7,7) כדי לקבל את הערכים הללו משתמשים בפקודות:

```
my @ar1=keys(%h); # @ar1 gets (1,"Zambura",5,"another")  
my @ar2=values(%h); # @ar2 gets ("elf",5,7,7)
```

למעשה, אם תנסו את זה באמת, סביר שלא תקבלו בדיוק את המערך שכתבתי, אלא שהאיברים יהיו בסדר אחר. זה בגלל שטבלת האש לא מאכסנת את האיברים בצורה סידורית, אלא לפי מפתחות. ואין דרך לדעת באיזה סדר הם יצאו:

כדי לגשת לאיבר או איברים מההאש לפי מפתחות:

```
$m=$h{"Zambura"};  
@ar=@h{1, "another"};
```

וכדי להכניס איבר חדש:

```
$h{"new one"}=3.14;
```

אגב, כמו תמיד בטבלאות האש, המפתחות חייבים להיות ייחודיים. זה אומר, שאם בדוגמא למעלה אני אכתוב "another" במקום "new one" אזי הערך הישן שתחת another יידרס (שזה 7) ובמקומו יהיה 3.14.

בשביל מה כל הסיפור הזה טוב? האש משמש להרבה דברים, כגון שמירה על מאפיינים שונים. אפשר לשמור מאפיינים שונים באותו האש, וכל מי שצריך מאפיין כלשהו, ניגש אליו ישירות באמצעות השם שלו, בלי לגעת במאפיינים אחרים. עוד שימוש זה לפברק מבני מערכים (struct של שפת C) כמה שזה נשמע מפתיע, לא קיימים בפרל מבנים, וכדי בכל זאת לפברק אותם פשוט משתמשים בהאש.

2.7. הקשרים ומשתנים

כל פעולה בפרל, מתבצעת תחת הקשר מסוים. הקשר יכול להיות סקלרי או רשימה, והוא נקבע על פי המשתנה אליו התוצאה צריכה להיות מוצבת. הנה דוגמא.

```
$acla=localtime();
$arr=localtime();
Thu " "
שתי הפקודות עושות את אותו הדבר, רק שהראשונה מחזירה מחרוזת של זמן, שנראית ככה: "Thu 13 Oct 1994 04:54:34" והשנייה מחזירה את המערך:
($sec, $min, $hour, $mday, $mon, $year, $yday, $isdst)
זה שימושי מאוד, כי ככה פונקציה יכולה להחזיר את התוצאה שמבוקשת, אבל שם רוצים יותר מזה, אז שמים אותה בהקשר מערך, והיא יכולה להחזיר יותר מערך יחיד.
למרות זאת, פונקציות בד"כ מחזירות תשובה מסוג יחיד, (סקלר או רשימה) בלי אפשרות לבקש משהו אחר. (אם כי פונקציות המנתחות מחרוזות כן נוטות לדואליות)
אם רוצים להכריח פונקציה להחזיר בהקשר סקלרי, משתמשים בפקודה scalar. דוגמא:
$arr=("Today", "is", scalar(localtime()));
אם לא היינו שמים את ה-scalar, אזי localtime היתה רואה הקשר רשימה, ושופכת את כל המערך הגדול שלה. ככה קיבלנו רק את המספר.
```

2.8. תוכנית דוגמא

מכיוון שעד כה כל מה שראינו היה סקלרים, מערכים והאשים, (אפילו לא מחרוזות!) אה, ואת הפקודה print, אז תסלח לי שהתוכנית קצת מטופשת. תתרגל.

```
#!/usr/bin/perl -w
use strict;
my ($a,$b,$c,$d)=(73,22,"Oops");
my (@arr1,@arr2,@arr3);
my (%h1,%h2);
@h1{"a","b","c"}=($a,$b,$c);
@arr1=keys(%h1);
@arr2=values(%h1);
@arr3=%h1;
push(@arr1,"klick");
push(@arr2,"klack");
%h2=("first"=>1,"second"=>2,"third"=>3,"forth"=>4);
$h2{"fifth"}=5;
splice(@arr3,0,scalar(@arr1),@arr1);
$h1{pop(@arr1)}=pop(@arr2);
$c=7;
$d=$a+$b*$c/$a;
print "And the number is: $d !\n";
print "Hash table: ", %h2, "\n";
print "Keys: ", @arr1, " Values: ", @arr2, "\n";
```

3. מבני בקרה

3.1. ביטויים בוליאניים

כמו שכבר הסברנו בפרק סוגי משתנים, סעיף משתנים סקלרים, בפרל 0, המחרוזת הריקה ("") ו-undef הם שקר, כל דבר אחר זה אמת. פרל תומכת בכל האופרטורים הבוליאניים המוכרים, אם בסגנון C, (&&, ||, !) או בסגנון פסקל. (And, Or, Not אופרטורים לוגיים :

פעולה	אופרטור פסקל	אופרטור C
And	$a \text{ and } b$	$a \ \&\& \ b$
Or	$a \text{ or } b$	$a \ \ b$
Not	$\text{not } a$	$! \ a$

אופרטורי השוואה : בפרל יש אופרטורים שונים להשוואה בין מספרים ולהשוואה בין מחרוזות. (טוב, בשפות אחרות כדי להשוות בין מחרוזות יש פונקציות מיוחדות)

פעולה	אופרטור מחרוזת	אופרטור מספר
האם שווה	$a \text{ eq } b$	$a == b$
האם לא שווה	$a \text{ ne } b$	$a != b$
האם גדול	$a \text{ gt } b$	$a > b$
האם גדול-שווה	$a \text{ ge } b$	$a >= b$
האם קטן	$a \text{ lt } b$	$a < b$
האם קטן-שווה	$a \text{ le } b$	$a <= b$
השוואה	$a \text{ cmp } b$	$a <=> b$

אופרטור ההשוואה מחזיר : 1 אם $a > b$, מחזיר 0 אם $a == b$, ומחזיר -1 אם $a < b$.

3.2. פקודת if וניגזרותיה

כן, ניגזרותיה. כלל גדול בפרל אומר, "יש יותר מדרך אחת לעשות את זה". ובפרט, יש יותר מדרך אחת לעשות if. אבל נתחיל ברגילה, בסדר? הנה:

```
if (condition1) {
  Set of commands...
} elsif (condition2) {
  Set of commands...
} else {
  Set of commands...
}
```

אלה שני משפטי תנאי רגילים, משורשרים.

יש גם וורסיות אחרות של משפט ה if-הפקודות הבאות זהות:

```
if ($m==3) {
  print "Avukado";
}
print "Avukado" if ($m==3);
print "Avukado" unless ($m != 3);
```

3.3. פקודת while

טוב, פה פרל זהה לשפת C. כמעט. הנה דוגמא:

```
while ($line=<>) {
  $count_lines++;
  if (substr($line,1) eq '#') {
    next;
  }
  if ($line eq "End\n") {
    last;
  }
  print $line;
}
```

אז מה יש לנו כאן? בשורה הראשונה יש את אופרטור ה-`<>`. זה אופרטור שקורא שורה מה-`stdin`, ומציב אותה לתוך `$line`. אם לא נשארו שורות - הוא יחזיר `undef`, מה שיתן שקר, והלולאה תסתיים. נסביר בפרק מאוחר יותר על האופרטור הזה.

מיד אחרי שקראתי את השורה, אני סופר אותה. שיהיה. אח"כ אני בודק האם התו הראשון של השורה הוא "#". אם כן, אזי אני מדלג ועובר לשורה הבאה. אם השורה היא "End", הסתיים הקובץ, ואני שובר החוצה מהלולאה. ובסוף, אם לא קרה שום דבר דרסטי, אני מדפיס את השורה לפלט הסטנדרטי (STDOUT).

3.4. לולאת for

לולאת for היא בהחלט כמו המקבילה של ב-C. כמעט.

```
my ($m, $i);
$m=1;
for ($i=1; $i<5; $i++) {
    $m = $m*$i;
}
```

יש ללולאות גם פקודות בקרה :

```
my ($m, $i, $j) = (0,0,0);
for ($i=1, $m=1; $i<5; $i++) {
    $j++;
    redo unless ($j>=$i);
    if ($i+$m>30) {
        last;
    }
    next if ($i==1);
    $m=$m*$i/$j;
}
```

פקודת next מקפיצה את ההרצה לסוף הלולאה, ומתחילה סיבוב חדש של הלולאה. כלומר, \$i מתקדם באחד, והלולאה שוב מתחילה לרוץ. פקודת last שוברת את הלולאה, והתוכנית ממשיכה לרוץ. פקודת redo היא פקודה מיוחדת. היא מקפיצה את ההרצה לתחילת הלולאה. זה אומר ש-\$i לא מתקדם, אבל \$j שממוקם שורה מתחתיו, כן יתקדם. לא ראיתי את זה בשום שפה אחרת, אבל דוגמא מעניינת להוראה תמצאו בפרק על טיפול בקבצים.

```
foreach $count (10,9,8,7,6,5,4,3,2,1, "BOOM!") {
    print "$count\n";
}
```

מקרה אחר של for היא ה-foreach. הנה דוגמא:

בכל סיבוב של הלולאה, \$count מקבל אחד מהערכים של הרשימה שאחריו, שמודפס בפקודה. אם הרשימה היא בעצם מערך, אזי \$count בעצם נהפך לערך. לא נתמקד במוזרות הזאת, רק נראה דוגמא:

```
my @arr=(1,2,3,4,5);
foreach $val (@arr) {
    $val=$val*2+1;
}
```

כאן הערכים בתוך המערך @arr ישתנו. מוזר, אך אמיתי ויעיל.

3.5 פעולות על תנאי

כשפרל בודק משפטים לוגיים, הוא עושה זאת בחלקים. לדוגמא, אם יש את הביטוי הבא:
`$x or $m++`
אם `$x` הוא אמת, אזי `$m` לא יקודם. זה בגלל שפרל כבר יודע שהערך של הביטוי הוא אמת, ואין לו צורך לבדוק את חלקו השני.

משתמשים בזה בפרל המון, פשוט במקום לעשות `if`. הנה פתיחת קובץ טיפוסית:
`open(my $fh, ">", "zorba.txt") or die "Cann't open that file!\n";`
הפקודה `open` פותחת קובץ, בשם `zorba.txt`, תחת המשתנה `$fh`. אם היא מצליחה, היא מחזירה אמת, ואז פרל לא ממשיכה לבדוק את התנאי. אם `open` מחזירה שקר, אז פרל ממשיך לבדוק את התנאי, ומגיע לפקודה `die "string"`, שמוציאה את הודעת השגיאה ועוצרת את ההרצה. נכון, זה יכול להיות מכווער, ואפשר לכתוב בעזרת זה דברים לחלוטין בלתי קריאים. אבל זה פרל, ואפשר להשתמש בזה במקום המון פקודות `if`. תגיד בעצמך מה יותר קריא, הפקודה שלמעלה או פקודה מקבילה בסגנון C?

```
if (open(FH, ">", "zorba.txt")==0) {  
    die "Cann't open that file!\n";  
}
```

בקיצור, אחרי שמתרגלים לזה, זה נחמד.

3.6 תוכנית דוגמא

כרגע יש לנו, כאבני בנין את המשתנים, ואת מבני הבקרה. התוכנית עכשיו יכולה להיות קצת יותר אינטיליגנטית. (אבל לא בהרבה)

```
#!/usr/bin/perl -w  
use strict;  
my ($key, $val);  
my %h;  
while ($key=<>) {  
    defined($val=<>) or die "Odd number of line?!\n";  
    chomp($key);  
    chomp($val);  
    $h{$key}=$val;  
}  
foreach $key (keys(%h)) {  
    print $key, "->", $h{$key}, ", ";  
}  
print "\n";  
print $h{"message"}, "\n";  
my $i ;  
for ($i=1; $i<$h{"loop count"}; $i++) {  
    last if ($h{"loop count"}*2-$i<5);  
    print $h{"loop msg"}, "\n";  
}
```

התוכנית משתמשת בפקודה שעדיין לא הזכרנו - `chomp`. זוהי פקודה שמורידה מהקלט את סימני סוף השורה. אם לא נשים אותה, התוכנית גם תעבוד, אבל נקבל כל מיני אזהרות. (בגלל שאנחנו משתמשים ב-`"-w"` בשורה הראשונה של התוכנית, שאומרת לו להזהיר אותנו) זה יקרה בכל פעם שננסה להשתמש בקלט כמספר, ואז הוא צריך לתרגם אותו למספר, ישמיט את סימני סוף השורה, יזהיר אותנו שהוא עשה את זה, אבל ימשיך לרוץ. מה שהתוכנית עושה זה לקרוא קובץ, וכל זוג שורות להכניס בתור מפתח וערך לתוך טבלת האש. אח"כ היא מדפיסה את טבלת ההאש שהתקבלה, מדפיסה את הערך שתחת המפתח "message", ואז רצה בלולאה לפי "loop count" ומדפיסה את "loop msg". קובץ קלט לדוגמא:

```
<Start input file>
5
Oops.
loop count
17
And the winner is
me.
Loop msg
AhAhAhAh!
message
Resistance is futile
<End input file>
```

שים לב שאין חשיבות לסדר הזוגות. ממילא הכל נשפך אח"כ לתוך האש ומתערבל שם. שים גם לב לכך שמספר השורות צריך להיות זוגי. אם הוא לא זוגי, (יכול להיות שהעורך שלך אוהב להוסיף שורה ריקה בסוף הקובץ) אז פשוט תוסיף בעצמך שורה ריקה, וכך יהיו לך שתי שורות ריקות בסוף הקובץ. הרצת התוכנית: אם נקרא לקובץ הקלט פשוט "in3", לקובץ של הסקריפט "chap3.pl", אזי ביוניקס השורה הבאה תפעיל את הקובץ:

```
./chap3.pl <in3
```

בחלונות, בהנחה שפרל היא בנתיב החיפוש:

```
perl chap3.pl <in3
```

בהצלחה.

4. טיפול במחרוזות

לפרל יש כוח עצום, בהשוואה לכל שפה אחרת, לטיפול במחרוזות. בעזרת האופרטורים והפקודות, קלות הפעולה בפרל היא ללא תחרות. רק לחשוב על כך שבפרל מחרוזת נחשבת סקלר, לעומת כל שפה אחרת, בה מחרוזת היא מערך או אובייקט, נותן איזו נקודה למחשבה.

4.1 אופרטורים של מחרוזות

```
my ($a, $b, $c) = ("Abra", "kadabra", 3);
print $a . $b, "\n"; # prints "Abrakadabra"
print $a x $c, "\n"; # prints "AbraAbraAbra"
```

4.2 מחרוזות ומשתנים

כדי לשלב משתנה בתוך מחרוזת, לא צריך פקודה מיוחדת. פשוט עושים את אחד מהשניים:

```
$temperature = 37;
$str = "The temp. today is $temperature degrees";
$str = "The temp. today is " . $temperature . " degrees";
```

באותה דרך, אפשר לשלב מערך, איבר המערך, כמה איברים ממערך, טבלת האש שלמה או חלק ממנה, וכו'. כל דבר שמתחיל ב-\$, @ או %, יכול להיכנס:

```
@temps=(23,53,25,34,11,7,0);
$str = "The temps in the whole week where @temps degrees, each day";
$str = "The temp. yesterday was $temps[6] degrees";
$str = "The temps yesterday and the day before were @temps[6,5]
degrees";
```

אם אתה רוצה להכניס את התווים \$, @ או % עצמם למחרוזת, פשוט שים \ לפניהם.

```
print "\$temp = $temperature\n";
```

אם אתה רוצה לעצב את המחרוזת בצורה מיוחדת, אתה תמיד יכול להשתמש בפקודה `sprintf`, המוכרת משפת C, ונתמכת בפרל.

4.3. ביטויים רגולריים

אוקי. לאט-לאט. תירגע. תנשום עמוק – ותתחיל.

מה זה ביטויים רגולריים? ביטויים רגולריים (Regular Expressions, regexp) הם דרך מסודרת לתאר מחרזות. אפשר גם לומר, שזו דרך לשאול שאלות על מחרזות. כמו שבמספר אתה יכול לשאול האם הוא גדול משלוש, במחרזות אתה יכול לשאול האם המחרזות מכילה "חתול". ללמד ביטויים רגולריים באופן מלא, גם ספר לא מספיק. ללמד באופן חלקי, צריך הסבר באורך גדול יותר מהחוברת הזאת. אני אתן כאן כמה עקרונות ודוגמאות, שיתנו לך את הבסיס לשימוש נורמאלי בכוח הזה שפרל נותנת לך. מומלץ להתעמק ולמצוא דוגמאות אחרות, כי פה טמון כלבלב גדול. אנחנו נעבוד בסעיף הזה בעזרת תוכנית דוגמא:

```
#!/usr/bin/perl -w
use strict;
my $str = "string";
print "Found!\n" if ($str =~ m/regexp/);
```

הדבר היחידי שאתה לא מכיר בתוכנית זה התנאי של ה-if. מה שאני שואל שם, זה האם הסקלר \$str מכיל את המחרזות "regexp". התשובה היא לא.

קודם כל – תנסה לשחק עם המחרזות, כדי לקבל אמת. (כלומר, תחליף גם את "string" וגם את "regexp", כדי לראות מתי אתה מקבל אמת ומתי שקר) קח לך עשר דקות. אני אחכה. עכשיו, נשנה את המחרזות למשהו מעניין יותר:

```
my $str = "Semuel casts: Abra Kadabra! sugoi"
```

מעכשיו בכל פעם נחליף בתוכנית את regexp לשאלה אחרת. הביטוי הרגולרי הפשוט ביותר, הוא המחרזות המבוקשת עצמה:

```
Abra Kadabra
```

כלומר, בתוכנית יהיה כתוב:

```
print "Found!\n" if ($str =~ m/Abra Kadabra/);
```

וזה יהיה אמת. המחרזות שלנו באמת מכילה "Abra Kadabra". עכשיו אפשר להתחיל לשחק:

הנקודה (.) מסמנת כל אות. אם למשל, במקום d לא אכפת לנו שיהיה J, 5, z או סימן כלשהו ^, (אות, מספר או סימן כלשהו) אנחנו יכולים להחליף אותה בנקודה:

```
$str =~ m/Abra Ka.abra/
```

המחרזות באמת מכילה את התת-מחרזות "Abra Ka", שאחריה יש תו כלשהו יחיד, (במקרה הזה יש את התו "d") ואחריה יש את התת-מחרזות "abra". לכן, יחזיר אמת. גם המקרים הבאים יחזירו אמת:

```
my $str = "Semuel casts: Abra kazabra sugoi";
my $str = "Abra ka5abra ahoi";
my $str = "blabLABlaAbra ka^abraapchiEWE!";
```

כמו שאתה רואה, לא חשוב לשאלה הזאת מה יש לפני המחרזות המבוקשת, מה יש אחריה, ואיזה תו בדיוק עומד במקום הנקודה. כל עוד המחרזות עומדת בתנאי, יחזיר אמת.

ננסה משהו אחר. נגיד שבמקום ה-b השני אנחנו רוצים שיהיה או b, או z או 5, ורק אלה. כלומר, נשים בתוך הסוגריים של ה-if את הביטוי הבא:

```
m/Abra Ka.a[bz5]ra/
```

יחזיר אמת. במחרזות באמת יש את התת-מחרזות "Abra Ka", אחריה יש תו כלשהו ("d") אחריו יש "a", אחריו יש אחד מהתווים "b", "z" או "5", ("b") ואחריו יש "ra".

רגע, בעצם התכוונתי, שיהיה כל דבר חוץ מ b, או z או 5:

```
Abra Ka.a[^bz5]ra
```

יחזיר שקר. כי במקום שאנו מתעקשים שלא יהיה "b", ולא "z" ולא "5", ויש "b". אי לכך, המחרזות לא תואמת לתנאי, ולכן שקר.

הסימן + מסמן אחד-או-יותר, על הדבר שלפניו. למשל, איפה שיש כרגע K, אני מוכן שיהיה יותר מ-K יחיד: (כלומר, לקבל גם את המחרזות "Abra KKKagakra")

Abra K+a.a[^\b5]ra
 בעצם, למה להגביל אותנו ל-K. נשלב בין שני הקודמים, ונסכים לקבל גם שבאותו מקום יוכל להיות גם T, M ו-R (כלומר, גם "Abra MTRTKTMRagakra" מתקבל)

Abra [KTMR]+a.a[^\b5]ra
 או סתם כל אות גדולה מה-ABC:

Abra [A-Z]+a.a[^\b5]ra
 הסימן * מסמל אפס או יותר. זוכרים את הנקודה שהכנסנו? אז בואו נעשה אותה אפס-או-יותר.
 כלומר, שם יכול להיות כל דבר שהוא, בכל אורך, או כלום:

Abra [A-Z]+a.*a[^\b5]ra
 זהירות עם הכוכבית. זה תיכלס יכול לבלוע לך קבצים שלמים. { איש אחד נכנס עם פיל לקולנוע. השומר אומר לו, אדוני, אתה לא יכול להיכנס עם פיל לקולנוע. האיש מסתובב והולך. אחרי חמש דקות הוא חוזר עם הפיל, ששתי פרוסות לחם תקועות לפיל באוזניים. השומר אומר לו, אדוני, אתה לא יכול להיכנס עם פיל לקולנוע. הוא עונה, אתה לא תגייד לי מה לשים בסנדוויץ' שלי! } אז תיזהרו שלא יהיה לכם פה סנדוויץ' פיל, בסדר?

הסימן כובע (^), בנוסף לפעילותו בתוך הסוגריים המרובעות, גם מסמן תחילת מחרוזת, או אם זוהי מחרוזת עם כמה שורות טקסט בפנים, אז תחילת שורה. בדוגמא הבאה, מחפשים משהו שמתחיל ב-
 Abra [A..Z]+a.a[^\b5]ra, אחריו לא משנה מה, שמסתיים בשורה חדשה ו-a[^\b5]ra.

Abra [A-Z]+a.*^a[^\b5]ra
 אגב, די פופולרי הצרוף הזה, ".*^". זה בגלל הקושי לסמן שורה חדשה, או שאתה לא בדיוק יודע מה יש בסוף השורה.

הנה סתם קשר 'אוי'. אני מוכן שיהיה באותו המקום במחרוזת או 'bra' או 'adr'.
 A(bra|adr) [A-Z]+a.*^a[^\b5]ra

בלי הסוגריים, זה עובד רק על אות בודדת: (בסוף, יש r או z)
 A(bra|adr) [A-Z]+a.*^a[^\b5]r|za

בעצם, אני לא בטוח בגלל שאני רוצה את ה-z והזאת בסוף. (כלומר, אפס-או-אחד)
 A(bra|adr) [A-Z]+a.*^a[^\b5](?r)a

טוב, די. אין לדבר הזה סוף. מה שקיבלנו כרגע, שאנחנו רוצים מחרוזת שמתחילה ב-A, אחריה יש bra או adr, אח"כ רווח, אח"כ אחת או יותר מהאותיות הגדולות באנגלית, (A..Z) אח"כ את האות a, אח"כ משהו, לא משנה כמה ומה. אח"כ שורה חדשה, a, ואז סימן כלשהו שיכול להיות כל דבר חוץ מ-z, b, או 5, ואז אולי r ואולי לא, ובסוף עוד a. ברור, לא? ☺

רק שים לב, שהמון תווים נחשבים לתווי בקרה בביטויים רגולריים, ואם אתם רוצים דווקא לחפש אותם, פשוט תשימו סלש-כפול (\) לפניהם. למה כפול? כי אל תשכחו שמה שאתם כותבים זה בסופו של דבר מחרוזת, ובמחרוזת סלש מרמז על תו מיוחד. סלש-כפול נהפך אחרי העיבוד של המחרוזת לסלש רגיל, שאומר לביטוי הרגולרי שפה מדובר בתו עצמו, ולא באיזה תו בקרה.

בנוסף לכל מה שאמרנו, יש גם דרכי קיצור לתיאור של תווים. במקום לכתוב [A-Za-z0-9], (שזה כל האותיות באנגלית, פלוס כל הספרות) אפשר להשתמש בקיצור \w, שאומר בדיוק את אותו הדבר. וזוהי טבלת הקיצורים:

צרוף	הגדרה	משמעות
\w	[a-zA-Z_0-9]	אות גדולה, קטנה, מספר או קו תחתי
\s	[\t\n\r\f]	רווח לבן (רווח, שורה חדשה, טאב וכו')
\d	[0-9]	סיפרה
\w	[^a-zA-Z_0-9]	הכול חוץ מאות מספר או קו תחתי
\s	[^\t\n\r\f]	הכול חוץ מרווח לבן
\D	[^0-9]	הכל חוץ מסיפרה
\b		קצה מילה מתקיים כאשר מצד אחד יש \s, ומהצד השני \S. מסמן קצה מילה.

ניתן כמה דוגמאות, בעזרת הפקודה split. הפקודה מקבלת מחרוזת וביטוי רגולרי, ושוברת את המחרוזת בכל מקום שהיא מצאה רצף המתאים לביטוי הרגולרי.

```
split("\\|", "start|middle|end"); # gets ("start", "middle", "end")
split("\\|d", "start|middle|end"); # gets
("start", "mi", "", "le", "en", "")
split("[td]+", "start|middle|end"); # gets ("s", "ar", "|mi", "le|en", "")
m/<a [^<]*href="[^"]*" [^>]*>[^<]*</a>/
```

האחרון הוא תנאי המחפש לינק של HTML בתוך מחרוזת. נשתמש בו בהמשך.

4.4 פקודת התאמת מחרוזת

הפקודה מחפשת תת-מחרוזת (המיוצגת ע"י ביטוי רגולרי) בתוך מחרוזת, ומחזירה את המחרוזת אם מוצאת. כמו כן, אפשר לבקש חתיכות מהמחרוזת. אשר להריץ את הפקודה שוב ושוב על אותה המחרוזת, כדי לקבל בכל פעם את תת המחרוזת הבאה. דוגמא:

```
while ($text=~m/Semuel is a(?n) (\w+) guy/g) {
  print $1, "\n";
}
```

הפקודה שבתוך התנאי של פקודת ה-while משמעותו, לחפש בתוך \$text את הביטוי המובא בין שני הסלשים. אם היא לא מצאה, היא תחזיר undef, שהוא שקר, והפקודה תסתיים. אם מצאה, היא תחזיר בכל פעם את המחרוזת שהיא מצאה, ומחרוזת מלאה זה אמת, שתבצע את הפקודה. כשיגמרו ההופעות של תת המחרוזת בתוך המחרוזת, היא תחזיר undef, ובא לציון גואל. שים לב לכך שבכל הקוד הוספתי פרמטר g לפקודה. זה אומר לו שיחפש בתוך המחרוזת שוב ושוב, עד שנגמרות לו התשובות, כמו שמוסבר בפסקה הקודמת. בלי g, הלולאה שלמעלה תרוץ לעד, שכל פעם היא מוצאת את אותו הדבר.

עכשיו, מאיפה צץ \$1?! שימו לב לסוגריים סביב ה-'w+'. הסוגריים אומרים, שאני מעוניין בקטע הזה של המחרוזת בנפרד. אז הערך מושם ב-\$1. אם יש יותר מסוגריים יחיד, אזי שמים את הבאים ב-\$2, \$3 וכו'. הסוגריים עם ה-n בפנים לא נחשבים, כי זה חלק מהביטוי הרגולרי. אגב, זו פקודה שמתנהגת שונה תחת הקשר רשימה. בהקשר רשימה, היא מחזירה רשימה של כל תת המחרוזות המתאימות לביטוי. דוגמא:

```
@arr=$text=~m/Semuel is a(?n) \w+ guy/g;
דוגמא ללולאה. אם $context מכיל קובץ html שלם, אז כדי לקבל את הלינקים שלו, עושים:
while($context=~m/<a [^<]*href="([^\"]*)" [^>]*>([^\<]*)</a>/g) {
  print "The url is: $1\n";
  print "The text is: $2\n";
}
```

תשאל, למה אני משתמש בכל מקום ב-'[^\<]'. במקום בסתם '*'. התשובה היא שהאלגוריתם שעובד פה הוא חמדן. זה אומר שאם אני אשתמש ב-'*', הוא ייקח מהלינק הראשון של הקובץ עד האחרון, תחת אחד ה-'*'. (סנדוויץ' פיל). ככה, אני מכריח אותו שלא לקחת את התווים '<>', מה שמגביל אותו שהוא לא יכול להתפשט ליותר מלינק אחד, כי זה מכריח שימוש ב-'<>'. אפשר להוסיף אופציות לפקודה, כגון האם להתחשב באותיות גדולות וקטנות ועוד. האופציות יתווספו אחרי הסלש השני של הפקודה. למשל:

```
$text=~m/abra/i;
```

4.5 פקודת החלפה

דומה לפקודת ההתאמה, רק שבמקום למצוא תת מחרוזת בתוך מחרוזת, פה אנחנו גם מחליפים את תת המחרוזת בביטוי אחר. דוגמא:

```
my $text="This is a long text, written slowing in the evening";
$text=~s/long/short/;
$text=~s/written \w+ (\w+)/scratched $1 fast/;
```

בפקודה הראשונה, אני פשוט מחליף את המילה "long" במילה "short". בפקודה השנייה, אני מחפש שלושה מילים, שהראשונה מהם היא "written", ומחליף אותם במילים "scratched fast", והמילה האחרונה מועברת לצד השני של הפקודה, ומשובצת שם. התוצאה תהיה:

```
"This is a short text, scratched in fast the evening"
```

4.6 חלק ממחרוזת

עכשיו נעזוב קצת את השדות הזוהרים של הביטויים הרגולריים. ניקח פקודות פשוטות, לא משהו מתוחכם - פשוט קבלת חלק ממחרוזת. בלי ביטויים רגולריים ובלי סיפורים. תן לי מהמחרוזת, מת מספר 5, שלושה תווים:

```
$newstr=substr($str,5,3);
```

מצד שני, אם נוסיף לפקודה עוד פרמטר, שהוא עוד מחרוזת, הפקודה תחתוך את הקטע מהמחרוזת, תכניס במקומו את הקטע החדש, (ואם הם באורכים שונים, היא תשנה את האורך של המחרוזת) ותחזיר את הקטע החתוך. תראה:

```
my $text="This is a long text, written slowing in the evening";
print $text, "\n";
print substr($text,5,5,"xxmmxxmmxx"), "\n";
print $text, "\n";
```

4.7 מציאת מיקום בתוך מחרוזת

אם אתה מחפש איפה האות g נמצאת בתוך מחרוזת, אז משתמשים בפקודת ה-index. דוגמא:

```
my ($str, $pos) = ("kenguru");
$pos = index($str,"g");
```

אפשר לעשות את זה על יותר מתו אחת, ואפשר גם להגיד לו מאיפה להתחיל לחפש. הנה קטע קוד פופולרי:

```
my $str = "Here we are today, there were the greens";
my ($pos,$old_pos)=(0,0);
while (($pos = index($str,"re",$old_pos))>=0) {
    print "Found \"re\" at $pos!\n";
    $old_pos = $pos+1;
}
```

הפונקציה rindex היא בדיוק אותו הדבר, רק שהיא מחפש מהסוף, ולא מההתחלה. אם תחליפו בקוד למעלה את index ב-rindex, (וכמובן, תשחקו עם הערכים ההתחלתיים של שני משתני המיקום) הוא יוציא בדיוק את אותו הפלט, רק בסדר הפוך. מהסוף, להתחלה.

4.8. חיבור מחרוזות

מי שקופץ ואומר, "אמרנו שיש אופרטור נקודה שמחבר מחרוזות!", אז צודק. זה טוב אם אנחנו רוצים לחבר שתי מחרוזות, ביחד. אבל אם יש לנו רשימה של מחרוזות, אז במקום לעשות for שבתוכו אופרטור הנקודה, יש לנו פקודה בשביל זה:

```
my @strarr=("Here", "we", "are", "today", "there", "were", "the", "greens");
print join('_', @strarr);
```

זה ידפיס: "Here_we_are_today_there_were_the_greens". שים לב לקו התחתי - זה מה שאמרת לי להכניס בין המחרוזות. זו הפקודה ההפוכה ל-split. שם חתכנו לפי איזה ביטוי, פה אנחנו מחברים בעזרת מחרוזת.

4.9. חיתוך שורה חדשה

אולי פעם קראת שורות מקובץ, וכל פעם טרחת לחתוך את סימני שורה-חדשה (n). פה, אתה פשוט משתמש בפקודת chomp:

```
my $line;
$line = <>;
chomp($line);
```

זוכר את האופרטור "<>"? הזכרנו אותו בעבר. זה אופרטור שקורא שורה מה-stdin. (שזה מהמקלדת, או מקובץ) אז אני לוקח פה שורה מה-stdin, ומכניס אותה ל-\$line. אולי יש בפנים את סימן שורה-חדשה, אולי לא. לא יודע.

chomp לוקח את המחרוזת, מחפש בפנים סימני שורה-חדשה, וחותר אותם. לא משנה אם אתה עובד על חלונות או על לינוקס. זה יעבוד. בכל פעם שתראה בתוכנית פרל קריאה מקובץ, תראה תמיד מיד אחריה את chomp. יעיל בטרוף.

4.10. סיכום מחרוזות

לא, אלה לא כל הפקודות של מחרוזות. יש המון. וגם לא תיארתי פה את כל מה שהפקודות מסוגלות לעשות. (למשל, לפקודה החלפה יש טונות של אופציות) אבל אלה הפקודות הבסיסיות ביותר והנפוצות ביותר, ואיתם תוכל לעשות משהו מועיל. וזה, בסופו של דבר, מטרת המסמך זה.

4.11 תוכנית דוגמא

יש לנו משתנים, לולאות ומחרוזות. למען האמת, אנחנו יכולים לכתוב עכשיו תוכניות סבירות בהחלט. למשל, תוכנית שעובדת על התגים של קובץ html :

```
#!/usr/bin/perl -w
use strict;
my @taglist=();
my ($line, $tag, $stag);
while ($line=<>) {
    chomp($line);
    while($line=~m/<([<>]*)>/g) {
        $tag=$1;
        $tag=~s/^\s+//;
        $tag=~s/\s+$//;
        ($stag)=(($tag=~m/\w+/));
        if (substr($tag,0,1) ne "/" ) {
            print "Starting tag($#taglist): $tag\n";
            push(@taglist, $tag);
        } else {
            while (($#taglist>0) and ($stag ne pop(@taglist))) {}
            print "Ending tag($#taglist): $tag\n";
        }
    }
}
```

התוכנית קוראת קובץ עם תגים, ועוקבת אחרי התחלה וסוף של כל תג. היא מתחילה בהגדרת ארבעה משתנים, ומתחילה לקרוא את הקלט שורה-שורה. עבור כל שורה, התוכנית מחפשת את התגים בשורה. (אל דאגה, התוכנית לא תיתקע על התג הראשון ותמצא אותו שוב ושוב, אלא בכל פעם שורת ה-while תחפש את התג הבא, עד שיגמרו התגים במחרוזת ואת הלולאה תסתיים) התג שנמצא מאוכסן בתוך \$1. (שים לב לסוגריים בתוך פקודת ההתאמה) אני מציב את הערך לתוך \$tag, מוריד רווחים בהתחלה ובסוף, ומכניס את המילה הראשונה בתג לתוך \$stag. אם אין את הסימן "/" בתחילת התג, זה אומר שמדובר בתחילת תג. אני מוציא הודעה מתאימה, הכוללת גם את גודל המערך שלי ואת התג עצמו, ודוחף את התג לתוך רשימת התגים שלי. אם יש את הסימן "/" בתחילת התג, זה אומר שזה סוף תג. אני מוריד מהמערך איברים עד שאני מגיע לתג המתאים, ומוציא שוב הודעה מקבילה על סוף תג. התוכנית מניחה שכל התגים הם באותיות קטנות, מכילים מילה אחת לפחות, ושהסגירה שלהם מתחילה באותה המילה. אם יהיה סגירה של תג בלי ההתחלה שלו, התוכנית תיגמר מאוד מהר. כמו כן התוכנית מניחה שאין שבירה של תג בין שתי שורות.

5. פונקציות והוראות

5.1. משתני ברירת מחדל

פרל עובדת דרך משתני ברירת מחדל. תכירו אותם: `$_`, `@_`, שהם המערך וסקלר ברירת המחדל. הכי טוב להראות דוגמא. ארבעת זוגות הפקודות הבאות זהות:

```
$m = shift(@_);  
$m = shift;  
print $_;  
print;  
@_=split($_);  
split;  
chomp($_);  
chomp;
```

כזכור, הפקודה `shift` לוקחת מערך ומורידה לו איבר בתחילתו. אם אתה לא אומר לפקודה לאיזה מערך אתה מתכוון, היא מניחה שאתה מדבר על `@_`. גם `print`, אם אתה לא אומר לה מה להדפיס, מדפיסה את `$_`. וגם `chomp`, כשלא אומרים לה לאיזה משתנה לבצע, היא מניחה שמתכוונים ל-`$_`. אבל `split` היא בכלל מקרה מיוחד. אם לא אומרים לה את מה לחתוך, היא מניחה שמתכוונים ל-`$_`. אם לא אומרים לה גם איפה לשים את המערך שהיא הרגע יצרה, היא שופכת אותו לתוך `@_`. (למעשה, ההתנהגות הזאת לא מומלצת, כי בטעות אתה עלול לדרוס בקלות את `@_`. אם תשתמש בה, תקבל אזהרה)

במבט ראשון זה מוזר. במבט שני זה מוזר. במבט שלישי, יש לזה פוטנציאל. אם כל הפונקציות והאופרטורים יעבדו ככה, לא נצטרך בכלל משתנים בקטעי קוד מסוימים. מצד שני כתבתי כבר אלפי שורות קוד, וחוף מהמקומות שמחייבים זאת, לא השתמשתי בהם. בתיעוד של כל פונקציה (לדוגמא: `perldoc -f shift`) כתוב בפרוטרוט מה הפונקציה רוצה לקבל, ומה יקרה אם היא לא תקבל את זה. לא כל הפונקציות משתמשות במשתני ברירת המחדל, לכן יש לעיין במדריך לצרכן לפני השימוש.

5.2. כתיבת פונקציות

פונקציה היא חתיכת קוד, המקבלת את הפרמטרים שלה בתוך המערך הדיפולטי, `@_`, ומחזיר לתוכנית הקוראת מערך או משתנה, (דרך המערך או המשתנה הדיפולטי) בעזרת הפקודה `return`. סימן הזיהוי של סוג הפונקציה שאני אראה כאן הוא התו `&`, אבל אף אחד לא משתמש בו כי אנחנו שמנים ועצלנים. (ופרל בד"כ מסוגל לנחש בעצמו שאתה מתכוון לקרוא לפונקציה בשם זה) השם הרשמי של הסוג הזה הוא `List-Operator`, זה כי היא מקבלת מערך בתור קלט.

דוגמא :

```
sub myChomp {
    $_=shift;
    chomp($_);
    return $_;
}
my $text="And the winner is\n";
print myChomp($text);
chomp($text);
print $text, "\n";
```

הפקודה shift מוציאה את האיבר הראשון מתוך @_ ושמה אותו בתוך \$_. הפקודה chomp, שכבר הכרנו אותה בפרק הקודם, בלי פרמטרים היא עובדת על \$_. ואז, אנו מחזירים את \$_ לתוכנית הקוראת. (תנסה גם להוסיף את התו '&' לפני myChomp בשורה השביעית, ותראה שפשוט לא תהיה לזה השפעה)

ננסה עוד דוגמא :

```
sub myTime {
    my ($msg1, $i, $msg2) = @_;
    print $msg1, (localtime)[$i]+1, $msg2, "\n";
}
```

אם נכתוב בגוף התוכנית:

```
myTime("Today is the ",6," day of the week");
```

נקבל: (בהנחה שמדובר ביום שלישי)

```
Today is the 3 day of the week
```

שים לב שפה לא החזרנו שום דבר. קורה. פונקציה לא חייבת להחזיר משהו. יכולה להחזיר מערך, סקלר, או כלום. ניקח שוב את הדוגמא האחרונה:

```
sub myTime {
    my ($msg1, $i, $msg2) = @_;
    print $msg1, (localtime)[$i]+1, $msg2, "\n";
    return ($msg1, (localtime)[$i]+1, $msg2);
}
```

פה היא מחזירה מערך, המכיל את שתי ההודעות שהתקבלו ואת הערך של localtime המבוקש. אבל אם נכתוב:

```
my $ans = myTime("Today is the ",6," day of the week");
```

התוצאה היא ש-\$ans=3. זה בגלל שהחזרנו מערך, ואם עושים סקלר שווה מערך, מקבלים בסקלר את גודל המערך. כדי שהפונקציה תדע מה התוכנית רוצה ממנה, משתמשים בשאלה \$wantarray. ככה:

```
sub myTime {
    my ($msg1, $i, $msg2) = @_;
    print $msg1, (localtime)[$i]+1, $msg2, "\n";
    if ($wantarray) {
        return ($msg1, (localtime)[$i]+1, $msg2);
    } else {
        return join(" ", $msg1, (localtime)[$i]+1, $msg2);
    }
}
```

אם רוצה מערך, נותנים לו מערך. אם רוצה סקלר, נותנים לו סקלר. בדיוק כמו ש-localtime יודע מה רוצים ממנו, ככה הפונקציה שלך.

הערה: בטח כבר הפנמת שהמשתנים עוברים דרך מערך. זה אומר, שבעצם אפשר להעביר רק מערך יחיד. הנה צורות קריאה נכונות ולא נכונות:

```
myFunc1($str, $i, @kaskas); #right.
myFunc2($str, @kaskas, $i); #wrong.
```

באפשרות השניה, כאשר תעשה בתחילת הפונקציה:

```
my ($str, @kaskas, $i) = @_;
```

אזי \$str יקבל את שלו, אבל @kaskas ישתה את שאר המשתנים, ו-\$i יקבל undef. לכן, ככה אפשר להעביר לפונקציה כמה סקלרים, אך רק מערך אחד, בסוף הרשימה.

אתה בטח עכשיו אומר, אבל יש פונקציות שמקבלות מערך בהתחלה, ואח"כ סקלרים ומערכים נוספים, כמו splice. זה נכון, אבל הם עובדים קצת אחרת. לא ניכנס לנושא, כי זה גורר נושאים אחרים כמו מצביעים. נסתפק בפונקציות רגילות, כאלה. ולסיום, משהו קטן: דרך פשוטה לעשות ערכי ברירת מחדל לפרמטרים:

```
sub write_person {
  my ($first, $last, $age, $address) = @_;
  $age ||= 23;
  $address ||= "the galaxy";
  my $msg = "$first $last is $age years old ";
  $msg .= "and lives in $address";
  return $msg;
}
print write_porson("John", "Rahul", 17), "\n";
```

הפונקציה רוצה לקבל ארבעה סקלרים - שני ערכים חובה, ושניים אופציונליים. כל ערך שלא יתקבל יהיה undef, ולכן יוכנס לו את הערך הדיפולטי.

5.3. הוראות לקומפילר

כשמריצים סקריפט פרל, הוא עובר שני שלבים: קומפילציה והרצה. אפשר לתת הוראות לקומפילר של פרל, ולהסביר לו מה אתה רוצה. לדוגמא, השורה מהתוכנית בפרק הראשון:

```
use strict;
```

הפקודה אומרת לקומפילר, שמפה והלאה הוא עובד במצב strict, מה שאומר כל מיני הגבלות. ההוראה תהיה תקפה עד סוף הבלוק בו היא נכתבה. נדגים על הוראה אחרת:

```
my $i;
for ($i=1; $i<=10; $i++) {
  use integer;
  print $i/3, "\n";
}
print "Final: ", $i/3, "\n";
```

ההוראה use Integer אומרת לקומפילר שמפה והלאה, להשתמש בקוד עבור מספרים שלמים. זאת אומרת, שהתוצאה תהיה 0,0,1,1,1,2,2,2,3,3. וההדפסה האחרונה תהיה 3.333.

כשאתה כותב use strict בתחילת התוכנית, אתה בעצם מחיל אותו על כל התוכנית. זה בגלל שהוא נכתב בהתחלה, מחוץ לכל בלוק. אם אתה רוצים לבטל את ההוראה לקטע מסוים, אזי תשתמש בפקודה no strict. לדוגמא:

```
no strict;
for (; $bla<28; $bla+=3) {
  print $bla, "\n";
}
```

אוטומטית יוגדר משתנה חדש, \$bla, יאותחל ל-undef, ויורץ בלולאה. בסיבוב הראשון נקבל שתי התראות שאנחנו עושים פעולות על משתנה לא מאותחל. אחרי הפעם הראשונה ש-\$bla מקודם בשלוש, הערך שלו יהיה שלוש, והלולאה תמשיך לרוץ. זה שימושי, אם אתה רוצה לעשות באיזה חלק תוכנית דברים שלא יעברו ב-strict, אבל יעשה לך חיים פשוטים. כמוכן שלא למדנו (ולא נלמד) במדריך הזה שום דבר כזה, אבל שתדע שהאופציה קיימת. הוראות נוספות במערכת:

```
use integer;
no integer;
```

מורה לקומפילר לעבוד עם מספרים שלמים בלבד.

```
use constant PI => 3.14;
print PI *10, "\n";
```

הגדרת קבוע. שים לב שכאשר משתמשים בקבוע, לא צריך את סימן ה-\$. אפשר גם להגדיר קבוע מערך:

```
use constant ARRAY => ( 1,2,3,4 );
print ARRAY;
print((ARRAY)[3]);
```

הגדרת הקבוע כופה הקשר רשימה מצד אחד, אך לא מערך. (תעיף מבט על ההסבר על המערכים בפרק 2, זה מוזכר) כלומר, כאשר הגדרנו את PI, בעצם הגדרנו רשימה שמה יש מספר אחד, והוא 3.14. ואז בפקודת ה-print הסתכלנו עליו בהקשר סקלרי, מה שנותן את האיבר האחרון (והיחיד, במקרה הזה) של הרשימה. מצד שני, כשרצינו להציג את האיבר השלישי של ARRAY, כדי לקבל מערך "אמיתי" היינו צריכים לכלוא את הקבוע בסוגריים. הצץ בעזרה של פרל:

```
perldoc constant
```

עבור עוד הרבה דוגמאות.

5.4 תוכנית דוגמא

זוהי תוכנית דוגמא, המקבלת רשימת מחרוזות ומחזירה אותם אחרי חיתוך הרווחים (או כל רווח-לבן) בהתחלה, הרווחים בסוף, וצמצום רווחים רצופים לרווח יחיד. כל זה, ללא שימוש בביטויים רגולריים.

```
#!/usr/bin/perl -w
use strict;
my @strs = ("first last", " somespace more space ", "\t\t
include\t\tabs\t \t as well\t \t", " ");
my @str = @strs;
foreach my $s (@str) {
    print "'$s'\n";
    $s = remove_leading_space($s);
    print "'$s'\n";
    $s = remove_trailing_space($s);
    print "'$s'\n";
    $s = reduce_space($s);
    print "'$s'\n";
    print "=====\n";
}
sub remove_leading_space {
    my $s = shift;
    my $i=0;
    my $l = length $s;
    while ($i < $l) {
        my $c = substr($s, $i, 1);
        last if (not is_white_space($c));
        $i++;
    }
    substr($s, 0, $i, '');
    return $s;
}
sub remove_trailing_space {
    my $s = shift;
    my $i=0;
    my $l = length $s;
    while ($i < $l) {
        my $c = substr($s, - $i -1, 1);
        last if (not is_white_space($c));
        $i++;
    }
    substr($s, -$i, $i, '');
    return $s;
}
sub reduce_space {
    my $s = shift;
    my $i = 0;
    while ($i < length($s)-1) {
        my $c = substr ($s, $i, 1);
        if (not is_white_space($c)) {
            $i++;
            next; # skip if not white space
        }
    }
}
```

```

    }
    $c = substr ($s, $i+1, 1);
    if (not is_white_space($c)) {
        substr($s, $i, 1, ' '); # make sure it is a space
        $i+=2;
        next; # skip two if next is not white space
    }
    substr($s, $i, 1, ''); # cut off one white space
    #$i++;
}
return $s;
}
sub is_white_space {
    $_=shift;
    return (($_ eq " ") || ($_ eq "\t") || ($_ eq "\f") || ($_ eq "\n")
|| ($_ eq "\r"));
}

```

הנה תוכנית שעושה את אותו הדבר, רק בעזרת פקודות החלפה.

```

my @str = @strs;
foreach my $s (@str) {
    print "'$s'\n";
    $s =~ s/^\s+//;
    print "'$s'\n";
    $s =~ s/\s+$//;
    print "'$s'\n";
    $s =~ s/\s+/ /g;
    print "'$s'\n";
    print "=====\n";
}

```

והנה תוכנית שלישית שעושה את אותו הדבר, גם בלי ביטויים רגולריים, תוך ניצול תכונה מיוחדת של פקודת split כשהיא פועלת עם פרמטר חיתוך רווח.

```

for my $s (@str) {
    $s = join ' ', split ' ', $s;
    print "'$s'\n";
}

```

6. טיפול בקבצים

6.1. הכוונת קלט\פלט

אחת הדרכים הקלות לטפל בקבצים, זה להזין אותם לתוכנית דרך הקלט\פלט הסטנדרטי. התוכנית תקרא את הקלט כאילו המשתמש כתב אותו, בעזרת האופרטור <>, ותוציא את הפלט בעזרת print רגיל, שינובת לקובץ אחר. דוגמא להרצת תוכנית כזו:

```
./myprog.pl <inFile >outFile
```

התוכנית תקבל את הקלט מתוך הקובץ inFile, ותוציא את הפלט לקובץ outFile. והיופי בזה, שהתוכנית בכלל לא יודע שהיא עושה את זה. מבחינתה, היא מקבלת הקלדות מהמשתמש, ומוציאה לו על המסך מידע.

אתה יכול לקחת את תוכנית הדוגמא שבסוף הפרק, להוריד את כל הוראות פתיחת וסגירת הקבצים, ופשוט לחתוך את המילים "InFile" ו-"OutFile". ואם נקרא לתוכנית LinkProcess.pl, אזי אפשר להריץ:

```
./LinkProcess.pl < urls.txt > urls.html
```

6.2. פתיחת קובץ

הכר את טיפוס הנתונים הרביעי של פרל: קובץ (File Handler). לטיפוס זה אין שום סימן לפניו, ובאופן מסורתי המשתנים נכתבים עם אותיות גדולות. כמו כן, סוג משתנה זה אינו דורש הגדרה, אלא אפשר פשוט להתחיל להשתמש בו, כאילו אין strict.

לפתיחת קובץ משתמשים בפקודת open. (מפתיע, אך אמיתי) אפשר לפתוח קובץ לקריאה ולכתיבה. ככה:

```
open(FH, ">", "wwwest.txt");
```

הפקודה הזאת פותחת את הקובץ wwwest.txt לכתיבה. אם הקובץ לא קיים, הוא נוצר.

בד"כ מוסיפים לפקודת open גם פקודת die:

```
open(FH, ">", "wwwest.txt") or die "Cann't open wwwest.txt!\n";
```

פקודת die היא ההתראה (Exception) של פרל. בניגוד לפקודת exit (שגם קיימת) שפשוט מפסיקה את הרצת התוכנית, die מוציאה הודעת שגיאה מסודרת, וסוגרת את התוכנית. אפשר לתפוס את סגירת התוכנית, ולהמשיך אותה בכל זאת, אבל זה סיפור אחר.

בכל מקרה, הפקודה שלמעלה מנסה לפתוח\ליצור את הקובץ לכתיבה. אם היא מצליחה, היא מחזירה אמת, ואין צורך להתקדם לפקודה הבאה. אם היא לא מצליחה, היא מחזירה שקר, ואז פקודת die פועלת ומפסיקה את ההרצה.

פתיחת קובץ לקריאה:

```
open(FH, "<", "wwwest.txt") or die "Cann't open wwwest.txt!\n";
```

פתיחת הקובץ להוספה (Append):

```
open(FH, ">>", "wwwest.txt") or die "Cann't open wwwest.txt!\n";
```

6.3. אופרטור <>

הזכרתי את האופרטור הזה בעבר. מה שהוא עושה, זה לקרוא מקובץ שורה. ככה:

```
$line = <FH>;
```

אם לא כתוב מחזיק-קובץ בין סימני הגדול-קטן, אזי הוא קורא מתוך ה-stdin, שזה הקלט הסטנדרטי. זה יכול להיות מהמקלדת, או מקובץ, תלוי איך מריצים את הקובץ. הנה דוגמה טיפוסית לקוד:

```
while (<>) {
    chomp;
    next if (m/^#/);
    if (m/\\s*$/) {
        s/\\s*$/ /;
        $_=$_.<>;
        redo;
    }
    print;
    print "\\n";
}
```

פקודת ה-while מפעילה את אופרטור ה-<>, שקורא שורה מתוך הקלט הסטנדרטי. מה ש-<> מחזיר זה תמיד אמת (כיוון שזה מחרוזת הכוללת בין השאר את סימן סוף השורה) אלא אם נגמר הקובץ, ואז זה שקר, מה שמפסיק את הרצת הלולאה. השורה שנקראה נכנסת למשתנה \$_.

שורה שניה - עושה chomp ל-\$_.

שורה שלישית - בודקת האם הקלט מתחיל בסימן '#'. אם כן - זוהי שורת הערה והולכים לשורה הבאה.

שורה רביעית - בודקים האם הקלט מסתיים בסימן '\'. (זה ביטוי רגולרי. האם המחרוזת מכילה '\', שאחריו אפס או יותר רווחים לבנים, שאחריהם סוף המחרוזת) אם כן, אזי צריך לחבר את השורה עם השורה הבאה. מוחקים את סימן ה-'\', קוראים את השורה הבאה, מחברים את השורות, וקופצים לתחילת הלולאה. (שים לב, זה redo. האופרטור <> שבתוך פקודת ה-while לא מופעל) ולבסוף, מדפיסים את \$_ וסוף שורה.

תריץ את הקוד הזה, ותנסה להכניס ידנית שורות שמתחילות ב-#, שורות שמסתיימות ב-\, וסתם שורות, ותראה מה קורה. לסיום, לחץ, Ctrl+Z, ואנטר.

אם תריץ את האופרטור <> בהקשר רשימה, אז הוא יקרא את כל השורות, ולא רק אחת.

```
my @lines;
@lines=<>;
```

אחרי זה, כל שורות הקלט יהיו בתוך המערך @lines.

6.4. כתיבה לקובץ

פה אין הרבה הפתעות - בשביל לכתוב לקובץ משתמשים בפקודה print. כמו שעד עכשיו כתבנו לתוך הפלט הסטנדרטי, ככה נכתוב לכל קובץ פתוח אחר:

```
open(FH, ">", "tmp.txt") or die "AHHHHH!\n";
print FH "The time is: ", scalar(localtime()), "\\n";
```

שים לב שאין פסיק בין FH לבין המחרוזת הראשונה. ככה זה. (ואם תשים פסיק, זו תהיה טעות. פקודת ה-print תנסה להדפיס את מחזיק הקובץ. דווקא חוסר הפסיק אומר לו שפה יש מחזיק קובץ, ולא משהו להדפסה)

6.5 סגירת קובץ

בצורה מפתיעה ביותר, בשביל לסגור קובץ משתמשים בפקודת `close`.

```
close(FH);
```

6.6 תוכנית דוגמא

```
#!/usr/bin/perl -w
use strict;
open(InFile, "<", "urls.txt") or die "Error IN!\n";
open(OutFile, ">", "urls.html") or die "Error OUT!\n";
print OutFile "<HTML><HEAD><TITLE>";
print OutFile "Semuel's Link Page";
print OutFile "</TITLE></HEAD><BODY>";
while(<InFile>){
    chomp;
    my $line=$_;
    if ($line=~m/^http/i) {
        print OutFile "<A HREF= $line > $line </A><br>\n";
    } else {
        print OutFile "$line <br>\n";
    }
}
print OutFile"</BODY></HTML>";
close(InFile);
close(OutFile);
```

התוכנית קוראת מקובץ טקסט המכיל רשימת לינקים, וכותבת קובץ html המכיל את אותה הרשימה, רק שעכשיו הלינקים הם באמת לינקים, (השורות שהם לא לינק עוברות דרך התוכנית, ואפילו מקבלות תג של סוף שורה בסופן) ואפשר להשתמש בקובץ דרך גלשן אינטרנט. (זה למעשה תוכנית ישנה שפעם עבדתי איתה. היה לי קובץ טקסט מסכן שם זרקתי את כל הלינקים שרציתי לשמור, והתוכנית הזאת הפכה את קובץ הטקסט למשהו שמיש)

7.1. כללי

עד עכשיו השתמשנו ביכולות הבנויות של פרל. אחלה יכולות, אבל מה הלאה? כדי להרחיב את היכולות של פרל, משתמשים במודולים. מודולים הם קבצי פרל, שאתה מוסיף לתוכנית שלך. זה כמו ה-include של C. מישוהו כבר ישב וכתב קוד המטפל בפעולות מסובכות. לא צריך להמציא את הגלגל מחדש בכל פעם. (למרות שזה כיף) קיימים אלפי מודולים עבור פרל. מודולים עבור תקשורת מול מסדי נתונים, מודולים עבור תקשורת מול האינטרנט, מודולים לניתוח קבצים, מודולים עבור מבני נתונים מסובכים. תבחר נושא, ובטח יש בשבילך מודול. כל המודולים נכתבו ע"י אנשים פרטיים, ולכן קיימים מודולים עם באגים, קיימים מודולים בהם המחבר נעלם, אבל רובם תקינים, והמחברים בד"כ מתקנים את הבאגים, אם וכאשר נמצאים כאלה. אז כנס לאתר www.cpan.com, ותרגיש חופשי למצוא מודול שעושה בדיוק את מה שאתה צריך.

7.2. שימוש במודולים

כדי להשתמש במודול, תשתמש בפקודה use. ככה:

```
use LWP::Simple;
    כאן אני מבקש מהקומפילר שיחבר לתוכנית מודול. המודול הספציפי הזה הוא לגישה לדפים באינטרנט. בתוך המודול כתוב איזה פונקציות ואילו משתנים הוא ייתן לי. למשל, המודול הזה נותן פונקציה הנקראת בפשטות get, שמקבל URL, ומחזירה את הדף שבאותו מיקום.
$content = get("http://www.google.com/");
    אחרי הפקודה הזאת, בתוך $content יהיה כל הדף הראשי של גוגל.
```

כאשר עושים use כמו למעלה, אזי המודול נותן לך איזה פונקציות שהוא רוצה. אבל יש פונקציות שהוא ייתן לך רק אם תבקש במיוחד:

```
use LWP::Simple qw{$ui};
    כאן אתה אומר לו, אני יודע שיש לך בפנים את $ui, תן לי אותו!
```

7.3. מודולים כאובייקטים

אין בפרל אובייקטים אמיתיים. במקום זה, משתמשים במודולים בתור אובייקטים. לדוגמא:

```
use FileHandle;
    זהו מודול שנותן גישה נוחה יותר לקבצים מאשר האופציות הפנימיות של פרל. (במיוחד אם אתה משתמש בכמה קבצים בו זמנית) כדי להגדיר באמצעותו מחזיק-קובץ, עושים:
$fh=new FileHandle;
    ואז, כדי לפתוח את הקובץ:
$fh->open("> tmp.txt");
    או פשוט:
$fh=new FileHandle("> tmp.txt");
```

שים לב רגע להבדל בין מודולים רגילים למודולים אובייקטים. מודולים רגילים מוסיפים לך עוד פונקציות לסביבת ההרצה, כמו בדוגמא שבסעיף הקודם. לפני שקראנו למודול לא היתה פקודת `get`, ואילו עכשיו יש. לעומת זאת, מודולים שהם אובייקטים לא מוסיפים לך פונקציות, אלא מחכים שתעשה להם `new` ודרך האובייקט תעבוד. כמובן, יש מודולים שעובדים בשני הדרכים, גם מוסיפים פונקציות וגם עובדים כמו אובייקט, אבל זו זכותם...

7.4. מודולים פופולריים

יש בעיות נפוצות יותר, ויש בעיות נפוצות פחות. בהתאם, יש מודולים שכולם משתמשים, ויש מודולים שבאים לתת מענה לבעיה ספציפית נידחת. הבאתי פה רשימה של המודולים היותר פופולריים, בהנחה שתזדקק להם בשלב זה או אחר.

שם מודול	תיאור
LWP::Simple	גישה לאינטרנט (בפרוטוקול http)
DBI	גישה למסדי נתונים
Mail::Box	גישה לתיבת הדואר (על יוניקס)
CGI	כתיבת תוכניות CGI
FileHandle	גישה נוחה לקבצים
Env	גישה למשתני הסביבה
Config	מידע על איך פרל קומפלה

7.5. תוכניות דוגמא

```
#!/usr/bin/perl -w
use strict;
use FileHandle;
my $handle = new FileHandle;
$handle->open("> output") or die "File won't open!\n";
$handle->print("A line in the river\n");
print $handle "And a line under the river\n";
seek($handle, 0, 0);
print "I printed: ", <$handle>;
$handle->close();
```

זוהי תוכנית המשתמש במודול `FileHandle`. אפשר לראות שמשתמשים במשתנה `$handle` גם בתור מחזיק קובץ רגיל, וגם כמו אובייקט. (כמו שאפשר לראות בשתי ההדפסות) עוד הערה, זה אנחנו פותחים את הקובץ באמצעות סימון מיוחד, `'<+>`. זה פתיחה עבור כתיבה וקריאה גם יחד. זה מאפשר לנו לעשות את פקודת ה-`seek` בתוכנית. התוכנית תפתח קובץ חדש (או תדרוס קובץ קיים) בשם `output`, תדפיס לתוכו שתי שורות, תקפץ חזרה לתחילת הקובץ, ותדפיס את מה שהודפס לתוכו.

```

#!/usr/bin/perl -w
use strict;
use LWP::Simple;
my $url="http://www.altavista.com/web/results?".
    "q=perl&kgs=0&kls=0&avkw=aapt";
my $i=1;
my ($content, $next_url);
while ($url) {
    $content=get($url);
    $i++;
    print $url . "\n";
    open(my $fh, ">", "altavista$i.html");
    print $fh $url, "\n";
    print $fh $content;
    close($fh);
    if ($content=~m|<a href="([\<>]+)"[\<>]*>\[Next &gt;&gt;\]</a>|) {
        $next_url=$1;
        $url="http://www.altavista.com".$next_url;
        sleep(10);
    } else {
        $url=undef;
    }
}

```

תוכנית המוציאה את כל הדפים שאלטא-ויסטא מוצאת על פרל. (היא שומרת את דפי אתר האלטא-ויסטא עצמם) אפשר לקחת את זה עוד צעד, ולהדפיס את ה-URL שנמצאו, (ע"י עוד פקודת while בסקריפט שמחפשת לינקים בתוך הדף) אבל זה תעשה בעצמך. התוכנית עובדת בעזרת LWP::Simple, המספקת לנו פקודת get, שמביאה דף מהאינטרנט. יש גם פקודת getstore, שמביאה דף מהאינטרנט ושומרת אותו לדיסק. יש עוד הרבה, אבל נסתפק בזה.

8.1. הקדמה

בראשית היה האינטרנט. ובאינטרנט היו שרתים, והשרתים שלחו ללקוחותיהם קבצים סטטיים מהדיסק הקשיח, ויהי ערב ויהי בוקר יום אחד. ויאמר העם: לא טוב לנו שהקבצים סטטיים. ויברא אלוהים את ה-Common Gate Interface, (או בקיצור CGI) ותשקוט הארץ שלוש שנים. ויתאנה העם על ה-CGI, כי איטי הוא, ואוכל משאבים. ויחר אף אלוהים, ויברא את ה-ASP ואת ה-JSP ואת ה-Templates, למען ענותם ולא יוכלו דבר עוד איש אל אחיו. וישב כל תכנת בנישה הנוחה לו, ויעסיקו עצמם במלחמות דת איש עם אחיו, ולאֵלוהים היתה המנוחה.

8.2. פרוטוקול HTTP על קצה המזלג

קודם כל יש את השרת, שמחכה ללקוחות. בצד השני יש את הלקוח, שרוצה איזה משאב מהשרת. הלקוח מתחבר לשרת, ושולח בקשה. הוא אומר, שלום שרת. אני לקוח מסוג (אינטרנט אקספלורר, מוזילה...) ומעוניין בקובץ (/users/list.html) הנמצא על שרת בשם (my.domain.com) והבקשה היא בשיטת (Get, Post). אני מכיר את השפות (אנגלית, עברית, ...) ותומך בעוד כמה דברים. (אתה לא חושב שאני אפרט הכל, נכון?) השרת מקבל את הבקשה, וחושב עליה. האם מותר למשתמש הני"ל לראות את המשאב הזה? האם המשאב הזה בכלל קיים? מאיפה אני משיג את המידע הדרוש? אוקי. עכשיו נגיד שההרשאות בסדר, והשרת מוכן להעביר את המידע ללקוח. רגע. לאט. קודם כל צריך לשלוח ללקוח מידע על מה הוא הולך לקבל. שלום לקוח יקר. בקשר לבקשתך, היא (מאושרת, אין לך הרשאות, לא קיימת, יש לי בעיה פנימית ואני לא יכול לספק לך את המידע כרגע). המידע הזה הוא באורך (מספר) בתים, ובקידוד (כלשהו). המידע לא השתנה מאז (אתמול, לפני שנתיים). עכשיו השרת שולח פעמיים אנטר ("ם") כדי לומר ללקוח שפה נגמרת הכותרת ומתחיל המידע האמיתי, ושולח את המידע. עכשיו לשיטות בקשה: יש שתי שיטות. (עיקריות) GET ו-POST. שיטת GET, אם פעם ראית כתובת בדפדפן שלך שנראית ככה: `http://my.domain.com/some/page.html?page=1&cat=normal` זה שיטת GET. כל הפרמטרים מופיעים פשוט בכתובת, כאשר סימן שאלה מסמן את תחילת הפרמטרים, ו- & מפריד בין פרמטרים. בשיטת POST אתה לא רואה את הפרמטרים בשורת הכתובת. הם מועברים בהחבא. היתרון הוא שזה נקי יותר, וכן אפשר להעלות קבצים בעזרת השיטה. מבחינת המתכנת שיושב וכותב את התוכנית בשרת, רוב הכלים המודרניים מחביאים את השיטה של הבקשה, ונותנים מנשק אחיד לקבלת המידע.

8.3. המודול CGI

זוהי הצורה הוותיקה ביותר לכתיבת תוכניות על השרת. בפשטות, מדובר בקובץ הרצה רגיל, שהשרת מריץ בכל פעם שהוא מתבקש לספק את המידע. השרת מאכיל מצד אחד את הסקריפט בנתונים, (דרך הסביבה, או דרך פרמטרים, או דרך ה-stdin) ומה שהסקריפט מוציא הוא שולח למשתמש. ומכיוון שאנחנו שמנים ועצלנים, יש לנו את המודול CGI שעושה את כל העבודה השחורה בשבילנו. דוגמא לסקריפט:

```
#!/usr/bin/perl -w
use strict;
use CGI;
use CGI::Carp 'fatalsToBrowser';

my $q = new CGI;
my $user_name = $q->remote_user();
my $filename = $q->param('filename');
print $q->header;
print "<html><head><title>Your file</title></head><body>
Hello $user_name !<br>
You have requested file: $filename !<br>
</body></html>";
```

זהו סקריפט CGI פשוט, המקבל את שם המשתמש מהזיהוי משתמש הבסיסי של השרת, מקבל בתור פרמטר (ולא משנה לנו איך הפרמטר הזה התקבל) שם של קובץ, ומדפיס למשתמש את השאילתה חזרה. שורה אחר שורה:

בהתחלה יש לנו את הקידומת הרגילה של כל תוכנית פרל, עם `use strict`. סטנדרטי. השורה `use CGI` נותנת לנו להפעיל את המודול, ואילו `use CGI::Carp 'fatalsToBrowser'` אומר לתוכנית, שאם יש שגיאה פאטלית, (התוכנית מסתיימת בצורה לא תקינה) אז שישלח הודעה יפה ומסודרת למשתמש, ולא משהו שהוא לא יכול לקרוא. השורה הבאה `my $q = new CGI;` יוצרת אובייקט חדש. כבר פה המודול משתלט על העניינים, קורא את כל הפרמטרים הרלוונטיים, משתני הסביבה וכל מה שהוא צריך, והוא כבר מוכן לעבודה. וכבר בשתי השורות הבאות אנחנו מפעילים אותו. הראשונה שואלת מה שם המשתמש שניגש לקובץ הזה (בהנחה שהזיהוי (Authentication) הבסיסי של השרת פועל) והשניה מקבלת פרמטר בשם `filename` שהמשתמש שלח עם הבקשה. עכשיו כשאני יודע מי המשתמש ומה הוא רוצה, אני יכול לומר לו את זה. אבל קודם כל – צריך לשלוח קידומת לתשובה. ואת זה עושה השורה `print $q->header`. שים לב ש-CGI לא מוציא מידע ישירות למשתמש, אלא רק מכין את התגובה הרלוונטית, ונותן לך לשלוח אותה. ככה הוא עובד. עכשיו כשגם את הקידומות שלחתי, הגיע הזמן לשלוח למשתמש את קובץ ה-HTML עצמו. סיימנו. רק שים לב שבתשובה יש גם את השם של המשתמש וגם את הקובץ שהוא רצה. את הקובץ עצמו לא שלחנו לו. שימשיך לחלום.

8.4. טפלייטים

כמו שראית בסעיף הקודם, ה-CGI פשוט מדפיס את קובץ ה-HTML. זה בסדר כל עוד מדובר במקרה פשוט, בו כל ההדפסה המיועדת נמצאת בסוף הקובץ. אם מדובר על מקרה מסובך יותר, בו צריך לחשב דברים תוך כדי הדפסה, יכול לצאת סקריפט מכוער מאוד, כאשר קטעי HTML וקטעי קוד מעורבבים בצורה מטורפת.

כדי להפריד את הקוד מה-HTML, הומצאו הטמפלייטים. בסגנון העבודה הזה, קובץ ה-HTML נשמר בנפרד, והקוד רץ בנפרד. ובתוך קובץ ה-HTML יש סימונים מיוחדים, שמתורגמים לפני שליחת הקובץ ללקוח. הדוגמא הבאה משתמשת בחבילה הנקראת `Template Toolkit`:

```
#!/usr/bin/perl -w
use strict;
use CGI;
use CGI::Carp 'fatalToBrowser';
use Template;

my $q = new CGI;
my $user_name = $q->remote_user();
my $filename = $q->param('filename');
print $q->header;

my $template_directory = "/home/semuel/tt";
my $config = {INCLUDE_PATH => $template_directory};
my $template = Template->new($config);

my $vars = {
    filename => $filename,
    user_name => $user_name,
};

# specify input filename, or file handle, text reference, etc.
my $input = 'print_that.html';

# process input template, substituting variables
$template->process($input, $vars) || die $template->error();
```

: והקובץ print_that.html מכיל

```
<html><head>
<title>Your file</title>
</head><body>
Hello [% user_name %] !<br>
You have requested file: [% filename %] !<br>
</body></html>
```

עכשיו נסביר. עד הקטע של הדפסת הכותרת, (print \$q->header) הכל בדיוק אותו הדבר. (חץ משורת ה-Template use)

שלושת השורות הבאות, זה סידור המנוע של הטמפלייט לקראת העבודה. קודם אני אומר לו איפה כל הטמפלייטים נמצאים, ואז מייצר אובייקט Template חדש.

אבל זה לא הכל. הטמפלייט צריך לדעת גם איפה נמצא קובץ ה-HTML, ומה להכניס לתוכו. את שם הקובץ הוא פשוט מקבל בתור פרמטר, ואת מה להכניס לתוכו הוא מקבל בצורת האש, המכיל זוגות שם וערך. ובסוף יש שורת process, שמקבלת את כל המידע ומדפיסה (בעצמה) את הפלט ללקוח. עכשיו אם תסתכל על הקובץ print_that.html, תראה משהו דומה מאוד לסוף של ה-CGI מהסעיף הקודם. רק במקום \$user_name יש [% user_name %]. המנוע של הטמפלייט בעצם מדפיס את כל הקובץ ללקוח, עד שהוא נעצר בסוגריים המרובעים הללו עם האחוזים. פה במקום להדפיס, הוא מסתכל בהאש שנתנו לו, מוצא מפתח שנקרא user_name, ומכניס במקום הסוגריים את הערך המתאים.

זו כמובן דוגמה פשוטה מאוד, המשלבת רק שני משתנים בתוך קובץ HTML. השפה של Template Toolkit היא מאוד רחבה, וכוללת משפטי תנאי, הכללת קבצים אחרים, ועוד הרבה דברים טובים. כמובן יש טמפלייט אחרים (הבדיחה טוענת שכל תכנת פרל כתב פעם מנוע טמפלייט משלו. למזלי אני ניצלתי מגורל זה. אני עצלן מדי. ©) שכל אחד עובד קצת שונה, אבל זה כבר מחקר נפרד. רק שים לב, שעדיין מדובר פה על תוכנית CGI. רק שבמקום להדפיס בעצמנו את ה-HTML נתנו למנוע טמפלייט לעשות זאת בשבילנו.

8.5 דפים משולבי קוד

בוא נסה לעבוד על הבעיה מכיוון אחר. במקום שתוכנית ה-CGI תדפיס את קובץ ה-HTML בחתיכות, בוא נהפוך את היוצרות. הקובץ עצמו יהיה קובץ ה-HTML, ובתוכו נשתול חתיכות קוד.

הגישה הזאת ממומשת, למשל, ע"י ASP. ברוב המקרים לא עובדים פה תחת CGI, אלא בסביבה אחרת שמראש יודעת שמדובר בקבצי HTML שבתוכם מושתל קוד. על סביבה כזאת נדבר בסעיף הבא. בנתיים, דוגמא:

```
<html><head>
<title>Your file</title>
</head><body>
Hello <% $r->connection->user() %> !<br>
You have requested file: <% $ARGS{filename} %> !<br>
</body></html>
```

זאת דוגמא הכתובה ב-Mason, שמהווה את הסביבה בה הקובץ הזה רץ. מכיוון שהסביבה מראש יודעת שמדובר בקובץ שכולל בתוכו פרל, היא מראש קוראת את כל הפרמטרים לתוך %ARGS ושולחת כותרת מתאימה. ועכשיו כמו במקרה עם הטמפלייט, הסביבה שולחת את הקובץ ללקוח, אבל את מה שתחום בתוך <% %> היא מריצה, ושולחת את התוצאה. גם פה אפשר לעשות את כל מה שמנוע טמפלייטים עושה, רק שהסגנון הוא יותר פרלי. הגישה הזאת מועדפת בד"כ ע"י מתכנתים שכותבים את האתר שלהם בעצמם, כיוון שהם כותבים את הלוגיקה בפרל בתוך ה-HTML. הגישה של הטמפלייט מועדפת בד"כ כאשר יש איש אחד שכותב את ה-HTML (ולא יודע פרל) ואיש אחר כותב את הפרל. ככה לזה שלא יודע פרל אין קטעי פרל שהוא לא מבין תקועים בתוך הקובץ.

8.6. ומה עוד

בסעיף השלישי הסברתי על CGI. בכל פעם שקובץ CGI מבוקש, השרת מריץ אותו ומדפיס את הפלט. הרצת תוכנית וסגירתה, כידוע, דורש הרבה ממערכת ההפעלה. היא צריכה לטעון פרל עצמו, לטעון את הקובץ, לקמפל אותו, להקצות לו זיכרון, להפעיל אותו ובסוף גם לסגור אותו. שלא לדבר אם הקובץ רוצה להתחבר למסדי נתונים ומשאבים אחרים.

מצד שני, היתרון הוא שהשרת לא צורך יותר זיכרון מהרגיל, (כי ה-CGI מקבל זיכרון משלו ממערכת ההפעלה) לא קשור רגשית אל ה-CGI, (אם קורת תקלה קולוסאלית, ה-CGI קורס באופן מרהיב, אך השרת לא אכפת לו. הוא רק משדר ללקוח שקרתה תקלה) ובכל פעם שה-CGI נסגר הוא לוקח עימו את כל הזיכרון שלו, כולל כל השגיאות והבעיות שיכלו לצוץ.

אבל בגלל שבכל זאת אנחנו רוצים לעבוד מהר יותר, המציאו את mod_perl. זה מודול של Apache, שנותן לו להריץ תוכניות פרל מבלי להצטרך לטעון אותם בכל פעם מחדש. יש אינטרפרטר אחד של פרל, שבכל פעם מקבל קובץ אחר להפעלה. הוא גם זוכר את הקבצים שהוא כבר הריץ, ככה שהשרת לא צריך לקמפל כל פעם מחדש את הסקריפט, אלא פשוט מריץ אותו שוב.

היתרונות: קודם כל מהירות. דבר שני, השרת יכול עכשיו לזכור דברים, כמו למשל אפשר לפתוח פעם אחת קישור למסד נתונים, ולהשתמש באותו הקישור לכל אורך חיי השרת.

בונוס נוסף: בגלל שפרל משולב עמוק לתוך Apache, אז יש גישה להרבה דברים שבתור CGI לא היה לנו. למשל, אפשרות לשנות את ה-url המבוקש, לפני שהשרת מתחיל ליצר את הפלט ללקוח. ועוד כל מיני דברים מתוחכמים.

החסרונות: צורך הרבה זיכרון. בגלל שפרל עכשיו הוא חלק מהשרת, השרת מחזיק אותו כל הזמן בזיכרון. מה שעושה אותו מנופח יותר, אפילו אם עכשיו הוא שולח איזה איקון ללקוח.

כמו כן, הוא זוכר דברים. שאלה אופיינית: למה השרת עושה <משהו> בפעם הראשונה שאני ניגש אליו, אבל בפעם השניה והלאה הוא עושה <משהו אחר>. התשובה היא כי יש לו משהו בזיכרון שנשאר מהפעם הראשונה.

לסיכום, שווה להשתמש בזה. יש המון יתרונות וחסרונות, אבל אם תכתוב את התוכנית כיאות, (ותתקל כמה פעם בבעיות, ותפתור אותם) ותשתמש בכמה טיפים שיחפו על החסרונות, רק תרוויח.

9.1. הקדמה

חסר לנו משהו בכל הסיפור. הממ... עברית! לעבוד באנגלית זה נחמד, אך אנחנו בישראל. אנחנו רוצים לקרוא, לכתוב ולעבד עברית! לרוע מזלנו, המחשב בד"כ עובד משמאל לימין. כדי לשכנע אותו לעבוד מימין לשמאל, דורש עבודה. כדי לשכנע אותו לעבוד דו-כיווני, זה בכלל מסוכן. אבל קודם כל אתה צריך כמה מושגי יסוד.

9.2. מושגי יסוד

בתחלה היה האסקיי. (ASCII) אסקיי זוהי טבלה הממפה מספר (מ-0 עד 127) לתו, ומכילה את כל הספרות, האותיות הגדולות והקטנות באנגלית, סימנים נפוצים, ועוד כמה תווים שונים. ואז בא אדם חכם (!) ואמר, בואו נוסיף פה עברית. טבלת האסקיי היא רק עד 127, כלומר רק 7 ביט, ולנו יש בכל בית (Byte) שמונה ביטים, כלומר עד 255, בואו נדחוף את האותיות העבריות מעל 127, והכל יהיה טוב. צצו עם זה שלוש בעיות: אחת, שגם הצרפתי, הגרמני, הספרדי והטיוואני חשבו ככה. וכולם שמו את האותיות שלהם באותו תחום. (128-255) כלומר, לאותו מספר היו כמה אפשרויות לתצוגה. הבעיה השנייה היא שבגלל שיהודים לא יכולים להסכים אפילו על השעה, היו כמה יהודים ששמו את האותיות העבריות בתחום ההפקר, וכל אחד שם אותם במיקום שונה. כלומר, לאותו תו היו כמה אפשרויות למספרים. הבעיה השלישית היא שיופי שלימדנו את המחשב את האותיות העבריות, עדיין נשאר לנו ללמד אותו בסופו של דבר, הטבלאות השונות הסתדרו במה שנקרא דפי-קידוד. (Code-Page) לכל שפה יש טבלה משלה, המקצה מספר לכל תו. כל דף מזוהה ע"י מספר. בעברית למשל יש את iso-8859-8 ואת CP1255. כל אחד משניהם מצה מקום לעברית, אבל במקום שונה מהדף השני. חסרון אחד ברור של השיטה הזאת הוא שאי אפשר להציג כמה שפות באותו המסמך. כי כל השפות דורכות אחת על השנייה. חסרון אחר זה שמישהו (מי?) צריך לומר למחשב באיזה דף קידוד אנחנו רוצים להשתמש.

9.3. יוניקוד

לעזרתנו בא איגוד היוניקוד. (Unicode) איגוד היוניקוד שם לו למטרה להקצות מספר אחד לכל אות שקיימת, ולא משנה באיזו שפה. כמובן שעבור זה צריך יותר משמונה ביטים. אז מה. האיגוד יצר טבלה ענקית, הכוללת את כל הסימנים בכל השפות. כמובן, שום דבר לא מושלם, ויש כמה מקרי קצה, (של תו שיש לו כמה מספרים, ומספר שיש לו כמה תווים) אבל בגדול, זה עושה את חיינו טובים יותר. עדיין, הטבלה הזאת לא מיתרגמת ישירות לדף-קידוד. וזה מהסיבה בפשוטה שהיא גדולה מדי. ולתו בודד יש רק שמונה ביטים, לא יותר. אז מה עושים? יש שתי כיוונים עיקריים: שיטה אחת אומרת: מהיום, לתו יש יותר משמונה ביטים. מהיום, יהיה לו 16 ביט. או אפילו 32 ביט. אלה שיטות הנקראות ucs2 ו-ucs4, בהתאמה. היתרון שלהם זה שלכל תו יש מספר ביטים קבוע. זה חשוב בטיפול בקבצים, שאם אתה רוצה לגשת לתו השלושים ושמונה, לא מעניין אותך מה יש לפניו. אתה פשוט קופץ לבית ה-38 כפול 2 או 4.

החיסרון הוא שלכל תו יש מספר ביטים קבוע. אם אני רוצה לכתוב רק אנגלית, אני עדיין מבזבז 4 בתים על כל אות, בעוד שבאסקי זה היה רק בית אחד על כל אות. שיטה אחרת אומרת: אנגלית נשארת בית אחד לכל תו. רק אם אני אצטרך, אני אוסיף בתים. ולזה נשתמש בביט השמיני של התו. אם הוא דלוק, (כלומר, ערכו 1) אזי אני מוסיף גם את הבית הבא לתו. אם הוא כבוי, אני נשאר עם בית אחד לתו אחד. שיטה זאת נקראת utf8. היתרון הוא בקומפקטיות, אם אני כותב אנגלית, יש לי בית אחד לכל תו. החיסרון הוא באורך תו משתנה. אני כבר לא יכול לדעת מראש איפה נמצא התו ה-38.

9.4. חזרה לפרל

כמו בהרבה מקרים, בפרל בחרו לשמר את הקיים. כלומר, מה שהיה עד כה, ימשיך לעבוד בלי שינוי. כלומר, אם התוכנית שלך כתובה באנגלית בלבד, וכך הקלט פלט שלה הוא באנגלית, הכל יעבוד כמו קודם.

יותר מזה. ברירת המחדל של פרל היא להניח שהתוכנית שלך כתובה בקידוד הנקרא latin1, שזה אנגלית פלוס סימנים נוספים. אם תנסה לדחוף כמה אותיות בעברית, אז בתוך מחרוזת זה כנראה יעבוד, מחוץ למחרוזת אלוהים יודע.

כדי לומר לפרל שאתה רוצה לעבוד יוניקוד בתוך הקוד, תכתוב:

```
use utf8;
```

ועכשיו אתה רק צריך לדאוג שהעורך שאתה משתמש בו באמת שומר את הקוד שלך בפורמט utf8. אחרת, מה עשינו?

כדי לומר לפרל שאנחנו רוצים לכתוב לקובץ ביוניקוד, יש את הפקודה binmode. ככה:

```
binmode(STDIN, "utf8");
```

וזה יגיד לפרל להתייחס לקלט שלו כמקודד ב-utf8. אפשר להפעיל את אותה הפקודה על הפלט ועל כל קובץ פתוח. כדי לפתוח קובץ עם קידוד כבר, אפשר לכתוב:

```
open(my $fh, "<:encoding(cp1255)", "file");
open(my $fh, ">:encoding(UTF-8)", "file");
open(my $fh, "<:utf8", "file"); # utf8 have it's own shortcut
```

הרעיון הוא להודיע לפרל באיזה קידוד המידע שהוא מקבל, והוא אוטומטית ימיר אותו לקידוד פנימי שלו (utf8). אני לא מגלה מה הפורמט הפנימי הזה (utf8), כדי שלא תסתמך עליו בהמרות אלא תודיע ספציפית לפרל מה הקידוד. בנוסף, צריך להודיע לפרל מה הקידוד הרצוי לפלט, והוא אוטומטית ימיר מהפורמט הפנימי הלא ידוע (utf8) לפורמט המבוקש.

כאשר פרל מקדד מחרוזות לפורמט הפנימי שלו, הוא מסמן את המחרוזות באמצעות דגל מיוחד. מרגע זה והלאה, כל פעולה המבוצעת על המחרוזות, עם זה חיתוך או ביטוי רגולרי, מודעת לכך שמדובר במחרוזת בה כל תו יכול להיות עם יותר מתו אחד, ומתייחסת בהתאם.

שים לב שהדגל הזה מדבק. אם תחבר שתי מחרוזות, אחת עם הדגל והשניה בלי, אז פרל ימיר את המחרוזת השניה לפורמט הפנימי שלו, ואז יחבר. אם היה לך במחרוזת השניה מידע שהוא לא אסקי, רוב הסיכויים שתקבל סימנים מוזרים.

אז תזכור - להודיע לפרל באיזה קידוד כל מחרוזת כמה שיותר מוקדם.

ואם אתה רוצה ידנית להעביר חתיכת טקסט מקידוד אחד לשני, יש מודול הנקרא Encode, שיעשה עבורך את העבודה:

```
use Encode qw{from_to encode decode};
# from iso-8859-8 to Perl's internal encoding (adding the flag)
$string = decode("iso-8859-8", $octets);
# from Perl's internal encoding to iso-8859-8 (removing the flag)
$octets = encode("iso-8859-8", $string);
# from legacy to utf-8 - "in-place"
from_to($data, "iso-8859-8", "utf8");
```

השורות לעיל ממירות לפורמט הפנימי של פרל ובחזרה לקידוד המקורי. בנוסף אפשר להמיר מידע מקידוד לקידוד, בלי ליצור מחרוזות ביניים, בעזרת הפקודה from_to. הפקודה, בניגוד ל-decode, אינה מפעילה את דגל הקידוד על המחרוזת.

כמובן, אפשר להמיר מכל קידוד לכל קידוד. כדי לקבל את רשימת הקידודים שהמודול הזה מכיר, עשה:

```
@all_encodings = Encode->encodings(":all");
```

9.5. זרימת טקסט

שים לב שלא דיברתי בכלל על העניין של כיוון הכתיבה. סיכמנו רק איך מייצגים תו בודד. במשך השנים קמו ונפלו כל מיני שיטות לבקרה על כיוון זרימת הטקסט. בסופו של דבר, השיטות מתחלקות לשתי ענפים: עברית ויזואלית ועברית לוגית. עברית ויזואלית טוענת, שאם המחשב מציג הפוך את המילים, אזי אנחנו פשוט נהפוך מראש את המילים. ואז, כאשר המחשב יציג אותם הפוך, אז הפוך-על-הפוך שווה ישר. (שיטה מאוד יהודית, אם יורשה לי להעיר) עברית לוגית טוענת, שאת המידע צריך לשמור ישר, ולא הפוך. תן לשכבות אחרות במערכת לעבוד עבורך ולהתיישר לבד. אנחנו בעד עברית לוגית. זה גם עוזר לעיבודים על הטקסט. כלומר, תחשוב למשל על מערך של מילים בעברית. בעברית ויזואלית האות הראשונה תהיה בסוף, ובעברית לוגית האות הראשונה תהיה בהתחלה. אם תריץ sort על המערך, באיזה משני המקרים המחשב יעשה מה שהתכוונת? בפרל אפשר לתרגם בין עברית ויזואלית לעברית לוגית ע"י:

```
use Locale::Hebrew qw/hebrewflip/;
```

```
$visual = Locale::Hebrew::hebrewflip($logical);
```

ובכן, נותרה עדיין השאלה של איך משלבים עברית ואנגלית באותה השורה. בעברית ויזואלית העניין למעשה די פשוט. ההנחה הבסיסית היא שהטקסט מוצג כאילו היה אנגלית, ולכן אפשר לחזות את ההתנהגות של הרכיב המציג. לעומת זאת, בעברית לוגית ההנחה היא שהרכיב המציג יודע איך לסדר את הטקסט. אבל איך הוא יסדר? לפי אילו כללים? נפתח פה פתח להמון אפשרויות. רק דוגמא קטנה:

```
Hello! How are you? לאומש!
```

למעלה המידע לפי סדר הבתים בזיכרון. עכשיו הרכיב המציג צריך לתרגם. שתי האפשרויות הבאות תקפות:

```
Hello! How are you? שמואל!
```

```
Hello! How are you? שמואל
```

האם הסימן קריאה שייך לטקסט באנגלית או לעברית? הרכיב המציג מחליט. גם פה, במשך השנים קמו ונפלו המון אלגוריתמים. גם ביוניקוד ניסו להכניס אלגוריתם גלובלי שיטפל בבעיה. רק מה לעשות שכל אחד מימש מהאלגוריתם של יוניקוד רק מה שנראה לו הגיוני. ושוב, חזרנו לאי תאימות הישנה והמוכרת.

בכל מקרה, יש מימוש חופשי של האלגוריתם של יוניקוד הנקרא FriBidi. אני מקווה שאם תצטרך תשתמש בו, ולא תמציא עוד אלגוריתם שלא תואם לשום דבר אחר.

9.6. זהירות – הגדרות

מעבר לכל מה שדיברנו, פרל יורש הגדרות ממערכת ההפעלה. זה אומר, שאם מערכת ההפעלה מכוונת ל-utf8, אז פרל יניח שהקלט והפלט הסטנדרטיים שלו הם utf8. (אבל לא קבצים שהוא קורא וכותב לדיסק, אגב. לא יודע למה) הקוד עצמו עדיין מוגדר להיות אסקיי טהור. בכל מקרה, אם אתה לא רוצה שיהיו אי נעימויות, ושהקוד יעבוד אותו הדבר בכל סביבה, ואם אתה עובד מחוץ לאסקיי טהור, אזי כדאי שתגיד מפורשות לפרל באיזה קידוד מגיע כל מידע, ובאיזה קידוד הקוד עצמו. אחרת, במוקדם או במאוחר יהיו צרות. מצד שני אתה יכול לספר לפרל שמדובר במידע בינארי, ולעשות מה שמתחשק לך, בזהירות.

9.7. ביטויים רגולריים

בסופו של דבר, זה מה שאנחנו רוצים לעשות. לנתח טקסט. אבל איך שואלים האם תו מסוים הוא בעברית? בשביל זה יש את מחלקות התווים. הנה:

10. תרבות וסביבה

ככלל, מתכנתי פרל הגיעו לשפה הזאת במקרה. גם אני. במהלך הלימודים לתואר עבדתי באיזו חברה, שם נתנו לי ספר על פרל, משימה וקדימה תתחיל לכתוב. מה הנקודה שלי? שרוב המתכנתים בשפה רואים בה כלי שימושי ואפילו כייפי, אבל לא ילכו לשבור למישהו את האף כי הוא אמר עליה דברים רעים. הפרק הזה מטרתו לעזור לך להתמצא בסביבה ולהבין כמה ביטויים נפוצים.

10.1. מלחמות דת

מלחמות דת (Religious Wars) הם וויכוחים עקרים שבד"כ מבוססים על העדפות אישיות. איזה עורך אתה מעדיף? vi או emacs? איזו שפה טובה יותר? ++C או Java? לדעתי זו תופעה סוציולוגית מרתקת, שעדיין לא היה לי הזמן והכוח לרדת לפשרה. אנשים מתקוטטים על שטויות, על שאלות חסרות תשובה ועל העדפות אישיות כאילו אין מחר. למרבה המזל, תכנתי פרל בד"כ לא נכנסים לוויכוחים אלו, כיוון שפרל זו השפה בה הם מתכנתים, ולא המזבח לו הם סוגדים. מצד אחד זה טוב, (הסביבה שקטה יותר והשכנים לא צועקים) מצד שני זה רע, כי המתכנת לא יגן על פרל ולא יפיץ את הבשורה ברבים. זה מה יש.

10.2. גישה

יש יותר מדרך אחת לעשות את זה. או בראשי תיבות: TIMTOWTDI. בד"כ, פרל נותנת ל שבע דרכים לעשות את אותה הפעולה. חלק מהדרכים איטיות יותר, חלק מהירות יותר. חלק יפות וחלק לא. הכלי הנכון למשימה. כמו שאמרת, תכנתי פרל הגיעו לפרל במקרה, ולכן אם יש לך שאלה, ומישהו חושב שזה רעיון טוב יותר לעשות את העבודה בכלי אחר או בשפה אחרת ולא בפרל, הוא יגיד את זה. כי פרל, כמה שאנחנו מספטים את השפה, היא רק עוד כלי. כלי חזק, אבל כלי. קרא את המדריך המזורגג - RTFM. העזרה של פרל היא די מקיפה. במקרה של שאלה פשוטה, לפעמים אנשים יעדיפו לשלוח אותך לקרוא את העזרה מאשר פשוט לענות על השאלה. ידוע גם כסינדרום "תן לאיש דג והשבעת אותו ליום. תן לו חכה ונתת לו מקצועי". כמוכן שאני נגד זה, ומעדיף לתת את התשובה ולמטה בהערה להפנות לעזרה לקריאה נוספת. כל אחד והגישה שלו. גולף. מכיוון שיש יותר מדרך אחת לעשות כל דבר, יש דרכים עם קוד קצר ויש עם קוד ארוך. לשחק גולף משמעו לחפש את התשובה הקצרה ביותר. ברוב המקרים במשחקים האלו נוצר קוד מכוער ביותר, כי כל מה שמעניין את המשחקים זה שהוא יהיה קצר, ולא שיהיה מובן. ולמה לבזבז שתי אותיות על סוגריים אם אפשר בלי?

10.3. שיתוף ורשיונות

מתכנתים הם עצלנים אך גאוותניים. שונאים לכתוב קוד שכבר מישהו כתב משהו דומה פעם, אבל אוהבים להראות קוד שהם כתבו לאחרים. כדי לעזור למתכנתים להשאר ככה, נולד האתר CPAN. המאגר האולטימטיבי של מודולים של פרל. תוכל למצוא שם אלפי מודולים, בכל נושא שבעולם, בהם אתה יכול להשתמש באופן חופשי, כדי להשאר עצלן אך מועיל. ואם כתבת משהו משלך, ואתה מעוניין לשתף את העולם בבשורה, אתה חופשי להעלות מודול משלך.

רגע, רשיונות. אם תשים לב, רוב המודולים מצהירים שהם מופצים תחת רשיון "זהה לפרל עצמה". הרשיון של פרל הוא אינו הרשיון של GNU. ב-GNU אסור לך לשלב את התוכנה בתוכנה מסחרית. בפרל מותר. בפרל גם מותר לעשות שינויים ולהפיץ אותה כחבילה חדשה, כל עוד אתה לא טוען שזאת החבילה הרשמית של פרל. הרעיון הוא שתוכל לעשות מה שאתה רוצה, כל עוד אתה לא פוגע בקהילה. (ע"י הפצת חבילות "מזויפות" וכד')

הרשיון הוא חלק חשוב ממה שעשה את השפה למה שהיא כיום. שב תקרא אותו, הוא לא נושך.

10.4. גרסאות של פרל

השפה מתפתחת במשך השנים. כיום, היא נמצאת בגרסא 5, ולא נדבר על גרסאות קודמות, כי זה ישן ודפוק מדי. אבל גם בתוך גרסא 5 יש גרסאות שונות. ואלו הן:

גרסא 5.004: הגרסא הראשונה בסדרה של 5. תמצא אותה בעיקר במחשבי יוניקס ישנים, עם מערכת הפעלה שלא שודרגה כבר הרבה זמן. התמיכה בגרסא הזאת הולכת ונעלמת.

גרסא 5.6: רוב המבנים המודרניים של השפה כבר קיימים. המינימום המומלץ לשימוש בפרל.

גרסא 5.8: פרל מודרנית. כוללת תמיכה ביוניקוד, ב-Threads.

גרסא 5.10: אין שינוי מהותי. הוסיפו משפט switch, (כמו ב-C) ושיפרו את מנגנון ה-Threading.

גרסא 6: פרויקט עתידי ושאפתני, בו מנסים לכתוב את השפה עצמה מחדש, להיפטר מכל מיני ירושות שפרל סוחבת מימי קדם ולהכניס תמיכה מסודרת באובייקטים. מעורבים בפרויקט הרבה שמות גדולים בעולם הפרל. (כמו לארי וול (Larry Wall), שהוא שכתב את פרל מלכתחילה)

11. מידע נוסף ועזרה

11.1. למה צריך עוד מידע

במסמך קטן זה ניסיתי לכלול את הבסיס של הבסיס של פרל. ניסיתי לכלול מספיק בשביל לתת לך, הקורא, כלים מספיקים עבור כתיבת תוכניות. אבל זה סתם מסמך. קיימים בפרל כל כך הרבה דברים שאפילו לא נגעתי בהם. אם אני אנסה אפילו לגעת בכל נושא, אפילו קצת, אני אסיים עם ספר בגודל מלא. בנוסף, גם נושאים שכתבתי עליהם, בד"כ לא כיסיתי הכל. למעשה, כיסיתי רק טיפה. אם תקרא את התיעוד של כל דבר שהזכרתי במסמך זה, על הפונקציה הכי זניחה שבעולם, אני מבטיח לך חגיגת שתמצא לפחות שלושה דברים שלא הזכרתי עליו. ספר אני לא יכול לכתוב, אבל אני יכול להפנות אותך לספרים ולמקורות שונים, בהם תוכל למצוא מידע נוסף, לקרוא על נושאים שונים, וללמוד.

11.2. ספרים

הספר הרשמי של עולם הפרל הוא Programming Perl, בהוצאת O'Reilly. (כרגע במהדורה השלישית) הספר הזה הוא אחד הספרים המקיפים ביותר על יסודות התכנות בפרל. והוא אפילו מצחיק. אם אתה צריך לקנות ספר אחד, זה הספר. הספר השני הוא Learning Perl, מאותה ההוצאה. נחשב גם לספר טוב, יותר בשביל מתחילים בעולם התכנות. מעולם לא קראתי אותו, כך שקשה לי להמליץ.

11.3. רשימות תפוצה

פרל מתברכת בקהילה גדולה של תכנתים, שישמחו לעזור לך בכל בעיה. למה שלא תצטרף לרשימת התפוצה הישראלית?

<http://www.perl.org.il/>

11.4. אתרים

11.5. ערוצי צ'אט באינטרנט

אה, כאן יש קצת בעיה. אנשי פרל בערוצי צ'אט, בד"כ חסרי סבלנות למתחילים. הם בד"כ ישלחו אותך לקרוא את התיעוד, במקום לתת תשובה. אז אם אתה מחפש מקום לברר איזה חלק של התיעוד לקרוא כדי לפתור בעיה, אולי זה באמת המקום לשאול אותה...

11.6. התיעוד

עם פרל מגיע תיעוד. זהו התיעוד העדכני ביותר לכל נושא ופונקציה. התיעוד כתוב מאוד מתומצת, ואם אתה מחפש דוגמאות, לא יהיו שם כאלה, בד"כ. מה שכן, כל אופציה הקיימת בפונקציה תהיה כתובה שם.

כדי להעלות תיעוד של פונקציה, למשל `chomp`, כתוב בשורת הפקודה:

```
perldoc -f chomp
```

שם תראה תיעוד מפורט על מה הפונקציה עושה, איזה פרמטרים היא מקבלת, מה היא מחזירה, מה קורה אם היא לא מקבלת פרמטרים, וכו'.

כדי להעלות תיעוד של מודול, כתוב בשורת הפקודה:

```
perldoc LWP::Simple
```

כדי לחפש מילים כתוב בשורת הפקודה:

```
perldoc -q keyword
```

ותקבל חזרה את רשימת המקומות בהם מילה זו נכתבה.

קיים גם אתר שמכיל את התיעוד של פרל עצמה, אם אתה (משום מה) מעדיף לקרוא דפים יפים באינטרנט במקום טקסט מחלון דוס:

<http://perldoc.perl.org/>

ובאתר של CPAN אפשר לקרוא את התיעוד של המודולים השונים.

11.7. פרקי קריאה

אם יום אחד יהיה לך הרבה זמן, המון ריכוז וטונות סבלנות, אתה מוזמן בזאת לשבת ולקרוא את הפרקים למטה מהתיעוד. כל אחד מהפרקים הללו מפרט נושא אחד בפרל, בצורה שלמה ומקיפה. (ומכוערת) אז כדי שתדע איפה לחפש, הנה רשימת הנושאים: (עשה לעצמך טובה, תנסה קודם לקרוא את המדריכים הנגמרים בסימון "tut", כמו `perlretut`, `perlreftut`. אלה חדשים יחסית, ומסבירים בצורה טובה יותר מהרגילים)

מילת קוד	נושא
<code>perl</code>	רשימת פרקי עזרה
<code>perlretut</code>	מדריך לביטויים רגולריים
<code>perlre</code>	ביטויים רגולריים
<code>perlvar</code>	משתנים מיוחדים בפרל
<code>perlsyn</code>	הסינטקס של פרל
<code>perllop</code>	אופרטורים
<code>perlsub</code>	פונקציות
<code>perlfunc</code>	הפונקציות הפנימיות של פרל
<code>perlreftut</code>	מצביעים
<code>perlrun</code>	פרמטרים עבור פרל עצמה. (לדוגמא, "w")
<code>perlmodlib</code>	הוראות לקומפילר ומודולים סטנדרטים
<code>perldiag</code>	הסבר על הודעות שגיאה

`perldoc subject`

`perldoc perl`

כדי להעלות הסבר על אחד מהנושאים כתוב :

כדי לקבל את הרשימה המלאה :

קריאה מהנה.

12.1. תודות

אני רוצה להודות :
לגבור, על המוטיבציה לכתוב מסמך זה.
לעופר קיי, על תיקון השגיאות הגסות שעשיתי במהלך הכתיבה.
לשלמה יונה, על תמיכה מורלית, ועל פרק העברית.
לעודד רזניק, מחברת RAZ מערכות, על הנספח "מלכודות למתכנת המתחיל".
לשאר חברי קבוצת דיון פרל ישראל, על אספקה שוטפת של בעיות ופתרונות.
לצה"ל, על שסיפק מהנדס משועמם עם המון זמן פנוי שיכול לשבת ולכתוב מדריכים.
לחברת Siftology מנוחתה עדן, שהכירה לי את עולם הפרל.
וכמובן, להורי, שהביאוני עד הלום. (-:

12.2. ביבליוגרפיה

O'Reilly's Programming Perl, 2nd edition.

12.3. על מסמך זה

מדריך זה נכתב ע"י שמואל פומברג.
המדורה ראשונה שוחררה בתאריך 20.9.2003, והמהדורה השניה והנוכחית שוחררה בתאריך 20.5.2008.
דף הבית שלו למציאת הגרסא העדכנית ביותר הוא :
<http://code.semuel.co.il/perlhebtut/>
ליצירת קשר, לעזרה ולתיקון שגיאות שמצאת במסמך, שלח מייל לכתובת :
owner@semuel.co.il

נספח א' – מלכודות למתחיל

יש כמה טעויות נפוצות של מתחילים. הנספח הזה הוא פשוט כותרת של טעות, וחתיכת קוד דוגמא שמראה מה נכון ומה לא, או פשוט מראה את ההבדל בין שני מקרים. אני משאיר לך לחשוב מה גורם להבדל בין שני המקרים. (טוב, חוץ מהכותרת שנותנת רמז "קטן")

ההבדל בין מחרוזות שתחומות ב-" (גרשיים) למחרוזות שתחומות ב-' (גרש בודד)

```
my $number = 57;
my $str1 = "There is a $number in the row";
my $str2 = 'There is a $number in the row';
print $str1, "\n";
print $str2, "\n";
```

ההבדל בין השוואת מספרים להשוואת מחרוזות.

```
print "Numerical!\n" if '5'=='05';
print "Strings!\n" if '5' eq '05';
```

קדימויות של אופרטורים לוגיים.

```
print "First!\n" if (0<1) or (2<1) && (3>4);
print "Second!\n" if (0<1) || (2<1) and (3>4);
```

משתנה ברירת המחדל \$_ הוא נוח, אך לא בטוח.

אין in במשפט foreach. (זו שגיאת קומפילציה)

```
my $i;
my @count = (1,2,3);
foreach $i (@count) { print "counting $i\n"; }
foreach $i in (@count) { print "not counting $i\n"; }
```

זהירות בשימוש באופרטור הטווח – אתה יכול להכנס ללולאות גדולות משהתכוונת.

```
my $i;
foreach $i ("a".."z") {print $i, " "; }
print "\n";
sleep(10);
foreach $i ("aaa".."zzz") {print $i, " "; }
print "\n";
```

בביטויים רגולריים, מכפילי התבנית (*,+) קשורים לתבנית שלפניהם.

נספח ב' - דברים נוספים שכדאי להכיר

כללי

בנספח זה מובאים דברים נוספים שטוב שתכיר, שיעשו לך חיים יותר קלים, או שתם אם תיתקל בזה בקוד של מישהו אחר, תדע על מה מדובר. הנושאים מובאים ללא שום סדר בעל משמעות.

גרשיים

ישנם הרבה סוגי גרשיים בפרל, כל גרש יכול להופיע ביותר מצורה אחת, ולכל גרש יש משמעות שונה. הנושא נמצא בעזרה תחת perlop, (אחרי כל האופרטורים) והנה טבלה שמסכמת את הנושא:

הגדרה רגילה	הגדרה גנרית	משמעות	אינטרפולציה
' '	q{ }	מחרוזת	לא
" "	qq{ }	מחרוזת	כן
` `	qx{ }	פקודת הרצה	*כן*
	qw{ }	רשימת מילים	לא
/ /	m{ }	חיפוש במחרוזת	*כן*
	qr{ }	ביטוי רגולרי	*כן*
	s{ }{ }	החלפה	*כן*
	tr{ }{ }	תרגום	לא
<<END		מסמך	*כן*

קודם כל, אני רוצה להצביע על כך שיש שתי דרכים לכל סוג של גרשיים: רגילה וגנרית. כלומר, לאו דווקא אתה חייב להשתמש בסמל הסטנדרטי. לדוגמה, שלושת השורות הבאות זהות:

```
$test = "A nice list \" with a '\"";  
$test = qq|A nice list \" with a '\"|;  
$test = qq|A nice list " with a '|;
```

השורה הראשונה היא סטנדרטית. מחרוזת עם גרשיים בתוכה, כאשר לפני הגרשיים יש סלאש, להודיע לקומפילר שהגרשיים הללו לא מסיימים את המחרוזת, אלא הם חלק מהמחרוזת עצמה. המחרוזת השניה היא כמו הראשונה, רק שאני משתמש בצורה הגנרית, ועובד עם הסימן "|" במקום גרשיים.

המחרוזת השלישית היא בדיוק כמו השניה, רק שפה הורדתי את הסלאש לפני הגרשיים שבתוך המחרוזת. מותר לי, כי אמרתי לפרל שהסימן שיסיים את המחרוזת הוא "|", ולכן זה לא הורס שיש גרשיים בתוך המחרוזת.

ככה לכל גרשיים יש צורה גנרית, שאם למשל יש לך הרבה '/' בתוך ביטוי רגולרי, במקום להשתמש בהרבה סלאשים, אתה פשוט בוחר בתו אחר:

```
$test = "A /line /with /alot /of /slashes and ddddd";  
if ($test =~ m|A /line /with /alot /of /slashes|) { print "Yes!"}
```

כל עוד אין | בתוך המחרוזת, לא צריך להוסיף הרבה סלאשים.
יש סוגי גרשיים שעושים אינטרפולציה, ויש שלא. אינטרפולציה זה לסרוק מחרוזת ולעדכן בה דברים.
דוגמא:

```
$type="rivar";  
$text="A line on the $type site";  
# $text gets "A line on the river site"  
זה אומר, שפרל סורק את המחרוזת, ומבצע בתוכה פעולות. אם שנשתמש בגרש יחיד:  
$text='A line on the $type site';  
אזי האינטרפולציה לא תתבצע, ומה שנקבל זה פשוט, "A line on the $type site". במחרוזת עצמה  
יהיה את הטקסט "$type", הוא לא יהפך למה שהשתנה $type מכיל, כי כאמור, אין פה  
אינטרפולציה.
```

איפה שכתוב שיש אינטרפולציה, אבל עם כוכבית, זה אומר שיש אינטרפולציה כל עוד לא השתמשת
בצורה הגנרית, ובתו גרש (!). אם השתמשת בגרש, אזי אין אינטרפולציה. דוגמא:

```
s/@uh/@oh/;  
s'@uh'@oh';  
בשורה הראשונה, פרל יחפש שני מערכים שנקראים @uh, @oh, ישפוך אותם לתוך תבנית החיפוש,  
ולפי זה יחפש. במקרה השני, פרל יחפש את המחרוזת "@uh" ויחליף אותה במחרוזת "@oh".  
ולתפקידי הגרשיים עצמם:
```

הסוג הראשון הוא גרש. כל תפקידו בחיים זה להחזיר את המחרוזת, בלי אינטרפולציה.
הסוג השני זה גרשיים. מחזיר את המחרוזת, אבל עושה לה אינטרפולציה לפני כן.
הסוג השלישי זה גרש-הפוך. הוא לוקח את הטקסט שבין הגרשיים, מריץ אותו בתור פקודה, ומחזיר
את הפלט של פקודה. מאוד שימושי. תנסה את השורה הבאה:

```
$wc=`wc *`;   
זה ייתן לך את מספר המילים בכל הקבצים שבספריה הנוכחית.  
הסוג הרביעי הוא רשימת-מילים. הוא מחזיר רשימת מחרוזות, כאשר בכל מחרוזת יש מילה אחת  
מתוך המחרוזת שניתנה לו. מעיין split. דוגמא:
```

```
@word=qw(one two three); # @word gets ("one", "two", "three")  
זה נפוץ בעיקר פקודות use שמגדירות מודולים. להזכיר, אם אתה רוצה שמודול יוציא משהו שהוא  
לא מוציא בד"כ, אתה כותב:
```

```
use LWP::Simple qw{$ui};  
ובמקום לתת רשימת מחרוזות, אתה משתמש ב-qw-כדי שהוא יצור אותה בשבילך. שים לב שגם אין  
פה אינטרפולציה, כדי שסימני הדולר לא יגרמו לו לעשות דברים לא רצויים.  
נקפוץ לסוג האחרון, ופשוט נראה דוגמא שלו:
```

```
print <<END  
A long text with  
new lines, # comments  
and such.  
END
```

מה שיקרה זה שפקודת ה-print תדפיס הכול, עד לנקודה שמופיעה המחרוזת END. אתה יכול לבחור
במקום END כל מחרוזת אחרת, ולהשתמש בה.

בלוקים

השורה הבאה תקפה :

```
$result=do {print "In da block\n"; 5};
```

מה קורה פה? ערך שווה לקוד בסוגרים מסולסלים?
בערך. מה שבתוך הסוגריים המסולסלים זה בלוק. הבלוק מורץ, והערך המוחזר הוא הערך שהפעולה האחרונה מחזירה, מה שבמקרה שלנו זה לא פעולה אלא מספר, 5.
לפעמים בלוקים משמשים כפונקציות אנונימיות. זה שימושי מאוד, להעביר לפונקציות פונקציות אחרות שעושות החלטות. נראה את פקודת sort :

```
my @arr=(1,9,4,24,52,3);
```

```
my @arr2;
```

```
@arr2=sort({$a cmp $b} @arr);
```

יש לי מערך מספרים, אני רוצה למיין אותו, אבל בסדר לקסיקוגרפי. (כלומר, 24 הוא לפני 4, בגלל שיש בהתחלה את הספרה 2) אם סתם הייתי מריץ את sort הוא היה ממיין לפי סדר מספרים. (4 לפני 24) אבל שלחתי לו פונקציה אנונימית, שעושה את ההחלטות בשבילי. וההחלטה היא למיין לפי סדר א"ב.

ככה אני יכול למיין מספרים לפי סדר בא"ב, לפי הערך האבסולוטי שלהם, לפי הסינוס שלהם, לפי הספרה האחרונה, איך שמתחשק לי. וכל מה שאני צריך, זה לשנות את הפעולות בתוך הסוגריים המסולסלים.

(הערה - שים לב למשתנים \$a, \$b בתוך הסוגריים המסולסלים. לא צריך להגדיר אותם עם my, אלא אלה גלובלים שפקודת sort משתמשת בהם. כמו כן, שים לב שאין פסיק אחרי הבלוק. ככה זה)

פקודת grep : זו פקודה שמקבלת מערך, בודקת איבר-איבר האם הוא מקיים איזה שהוא תנאי. אם כן, מעבירה אותו למערך שהיא מחזירה. אם לא, היא חוסמת אותו. מעיין מסננת של מערכים. הנה דוגמא :

```
@arr2=grep({$_>7} @arr);
```

הפקודה תחזיר את אברי מערך @arr הגדולים משבע, שזה 9,24,52. השאר לא יוחזרו.

הנה דוגמא שבודקת האם משתנה מסוים מכיל אחד מכמה ערכים :

```
if (grep({$_ eq $c} ('\n', '\t', ' ', '\r')) {  
    print "equal!\n";  
}
```

```
}
```

זה מעיין grep בהפוך. אנחנו מסננים מערך של איברים קבועים מול איזה תו המגולם ב-\$c. אם התו הוא אחד מהם, אזי מוחזר מערך עם איבר יחיד, שבהקשר סקלרי זה הערך '1', שזה אמת. מסובך להבנה בהתחלה, אני מודה, אבל נחמד.

lvalue

התרגום הטוב ביותר שאני יכול למצוא ל-lvalue הוא "ערך הניתן להצבה". לדוגמה \$c הוא כזה, בעוד ש-5 הוא אינו כזה. אם אתה יכול להציב לתוכו ערך, אזי הוא lvalue. אם לא, אז לא. דוגמאות:

ערך	האם lvalue
\$c	כן
\$arr[5]	כן
@arr[5,2]	כן
@h{'one', 'two', 'three'}	כן
(\$c>5)	לא
split('^',\$str)	לא
splice(@arr,2,3)	כן
substr(\$str,4,2)	כן

בשביל לוודא שמדובר ב-lvalue, כל מה שאתה צריך לעשות זה לנסות לשים אותו מצידו השמאלי של סימן שווה, ולראות האם התוכנית רצה. אם כן, זה lvalue. אם לא, אז לא. הנה, דוגמא:

```
my $str="And now the show begins";  
substr($str,4,2)="aVuKaDo";  
print $str, "\n";
```

פה מוגדרת מחרוזת התחלתית, ואז אני משנה חתיכה ממנה למשהו אחר, מה שגורם לה להתרחב ולאבד כל הגיון. (אבל זו בעיה אחרת)

מרחבי שמות

קודם כל, אני רוצה להציג לפניך את העובדה המהממת שמערך וסקלר שונים יכולים להיות בעלי אותו השם. יכול להיות ששמת לב לזה, בפרק חמש, כשהצגתי את המערך והסקלר הדיפולטי, ששניהם השם שלהם הוא קו-תחתי, אבל זה תקף לכל שם אחר גם כן. כלומר, יכול להיות משתנה \$temp ומערך @temp וטבלת האש %temp ופונקציה &temp ומחזיק-קובץ temp, כולם קיימים תקפים ונגישים בעת ובעונה אחת. (ההסבר הבא נכון למשתנים המוגדרים בעזרת our, שהם גלובלים ונגישים מכל נקודה בתוכנית. משתנים המוגדרים בעזרת my אינם גלובלים אלא מקושרים לקוד, ומנגנון הגישה אליהם שונה) מה שקורה זה שיש מרחב שמות, נגיד עכשיו main. המרחב הוא בעצם טבלת האש, עם שמות של משתנים בתור מפתח, ווקטור קטן בתור ערך. (אפשר לעשות את זה) הווקטור מכיל חמישה ערכים - אחד בשביל הסקלר בשם זה, אחד בשביל המערך, וכן בשביל טבלת ההאש, הפונקציה ומחזיק הקובץ. איך פרל מבחין בין כל המשתנים עם אותו השם? לפי ההקשר. \$temp שונה לגמרי מ-\$temp[3] ומ-\$temp{"hello"}. כולם מתחילים ב-\$temp, אבל אחד הוא המשתנה הסקלרי בשם זה, השני הוא האיבר השלישי במערך @temp, והשלישי הוא איבר בטבלת ההאש %temp שהמפתח שלו הוא .hello ועכשיו למרחבי השמות עצמם. התוכנית הראשית נמצאת במרחב שמות הנקרא main. שלושת המשתנים הבאים הם בעצם אותו המשתנה:

```
$temp  
$::temp  
$main::temp
```

שלושתם הוא המשתנה \$temp במרחב השמות הראשי.

לכל מודול ואובייקט, יש מרחב שמות משלו. למשל, אם אתה משתמש במודול LWP::Simple, ורוצה לגשת למשתנה \$ui המוחבא, פשוט תכתוב \$LWP::Simple::ui. כשאתה משתמש במודול, נפתח בשביל המודול מרחב שמות משלו מצד אחד, והוא שופך לתוך מרחב השמות שלך את כל המשתנים שאתה צריך כדי להשתמש בו מצד שני. ככה יש הפרדה בין המשתנים הפרטיים של המודול למשתנים החיצוניים. אבל פרל לא מגבילה אותך מלגשת למשתנים של המודול. אם אתה באמת רוצה, בבקשה.

נספח ג' - רשימת פונקציות מובנות

רשימת הפונקציות בפרל, לפי סדר האלפבית. (הלועזי, כמובן)

שם פונקציה	קטגוריה	תיאור
abs	מתמטיקה	הערך האבסולוטי של הפרמטר
accept	רשת	קבלת קשר (tcp וכאלה) על הרשת
alarm		שבירת המתנה. אם אתה הולך לחכות למהשהו, ולא רוצה לחכות לעד, אתה יכול להשתמש בזה כדי לצאת מהמתנה פעם בכמה זמן
atan2	מתמטיקה	arctan. טריגונומטריה. (אגב, אין את פונקציה ה-tan בפרל. תשתמש ב-cos ו-sin)
bind	רשת	אחד מהשליבים בפתיחת קשר על הרשת
binmode	קבצים	טיפול בקבצים בינאריים
bless		שלב ביצירת אובייקט
caller		בעזרת פקודה זו, פונקציה יכולה לברר מי הפעיל אותה
chdir	קבצים	לעבור לספריה אחרת
chmod	קבצים	לשנות הרשאות של קבצים
chomp	מחרוזות	חיתוך סימני סוף שורה ממחרוזת
chop	מחרוזות	חיתוך התו האחרון ממחרוזת (עדיף chomp)
chown	קבצים	לשנות את הבעלות על קבצים
chr	מחרוזות	מחזיר את התו עם ערך האסקי הנתון
chroot	קבצים	חוסם גישה של התוכנית לקבצים הנמצאים מחוץ לספריה מסוימת
close	קבצים	סוגר קובץ פתוח
closedir	קבצים	סוגר ספריה פתוחה
connect	רשת	אחד השליבים בפתיחת קישור על הרשת
cos	מתמטיקה	מה קרה, לא למדת טריגונומטריה!?
crypt		הצפנת סיסמת הכניסה ביוניקס
dbmclose	מסד נתונים	מפריד בין טבלת האש למסד נתונים פשוט (ראה dbmopen)
dbmopen	מסד נתונים	מקשר בין טבלת האש למסד נתונים פשוט (ראה dbmclose)
defined	משתנים	בודק האם משתנה הוא undef או לא
delete	טבלאות האש	מוחק זוג מפתח/ערך מטבלת ההאש
die		עוצר את פעולת התוכנית, ומדפיס את הודעת השגיאה. (אלא אם אתה עובד תחת eval)

שם פונקציה	קטגוריה	תיאור
do		יכול לשמש ביחד עם while, "do { ... } while (condition);" או כדי להריץ קובץ פרל: do 'int.pl';
dump		אה, פונקציה לא שימושית.
each	טבלאות האש	מחזיר זוגות מפתח-ערך מההאש בכל פעם שהוא נקרא. ככה אפשר לעבור על ההאש ערך אחרי ערך.
eof	קבצים	מחווה האם קובץ הפתוח לקריאה הגיע לסופו
eval		מריץ קטע קוד בצורה "בטוחה", כך שאם die הופעלה, התוכנית לא תסתיים
exec		מריץ תוכנית חיצונית
exists	טבלאות האש	בודק האם מפתח מסוים נמצא בטבלת האש
exit		עוצר את פעולת התוכנית. נקודה.
exp	מתמטיקה	אקספוננט. e בחזקת משהו. כאילו, דה!
fcntl	קבצים	משנה תכונות של קבצים
fileno		נותן לך את מספר הקובץ, כנהוג ב-C, למחזיק הקובץ שלך
flock	קבצים	נעילה של קבצים, לקריאה או לכתובה
fork		מפרק את התוכנית שלך לשתי תוכניות שונות
format	קלטופלט	משהו מסובך להוצאת פלט
formline	קלטופלט	כניל
getc	קלטופלט	מקבל תו יחיד מהקלט
getgrent		מחזיר את הקבוצות משתמשים במערכת
getgrgid		מחזיר קבוצת משתמשים במערכת, לפי המספר שלה
getgrnam		מחזיר קבוצת משתמשים במערכת, לפי השם שלה
gethostbyaddr	רשת	מחזיר נתונים על מחשב באינטרנט לפי כתובת ה-IP שלו
gethostbyname	רשת	מחזיר נתונים על מחשב באינטרנט לפי השם שלו
gethostent	רשת	מחזיר נתונים על מחשבים שרשומים אצלך, בקובץ ה-hosts
getlogin		נותן את שם המשתמש
getnetbyaddr	רשת	לא מעניין
getnetbyname	רשת	באמת לא מעניין
getnetent	רשת	ממש לא מעניין
getpeername	רשת	מחזיר את כתובת ה-IP של המחשב בצד השני של הקשר
getgrp		נותן באיזה קבוצת תהליכים נמצא תהליך מסויים
getppid		נותן את מספר התהליך של תהליך האב של התוכנית שלך
getpriority		באיזו קדימות התהליך הזה? (או תהליך אחר)

שם פונקציה	קטגוריה	תיאור
getprotobyname	רשת	
getprotoent	רשת	
getpwent		
getpwnam		
getpwuid		
getservbyname	רשת	
getservbyport	רשת	
getservent		
getsockname	רשת	מחזיר את הכתובת שלך עבור קישור רשת נתון
getsockopt	רשת	מחזיר את הפרמטרים עבור קישור רשת נתון
glob	קבצים	מחזיר שמות של קבצים עבור שמות גנריים. לדוגמא, '*.c' יהפוך לרשימת כל הקבצים עם סיומת 'c'.
gmtime		מתרגם זמן מקומי לזמן גרינוויצ'י
goto		כל שפה צריכה goto בפנים, לא?
grep	מערכים	סינון רשימה לפי תנאי. ראה נספח ב'
hex	מתמטיקה	מתרגם מחרוזות המכילה מספר הקסה-דצימלי לערך
import	מודולים	מבקש ממודול לשפוך את הפונקציות שלו לתוך מרחב השמות שלך. אם זה לא מובן לך, אז אתה עדיין לא צריך להכיר פונקציה זאת
index	מחרוזות	מוצא את ההופעה הראשונה של תת מחרוזות בתוך מחרוזות
int	מתמטיקה	מחזיר את החלק השלם של המספר
ioctl	קלטופלט	הגדרות ברמה נמוכה של אמצעי הקלטופלט
join	מחרוזות	מחבר רשימת מחרוזות למחרוזת אחת, ושם ביניהם תת-מחרוזות נתונה.
keys	טבלאות האש	מחזיר את רשימת המפתחות הנמצאים בתוך טבלת האש
kill	תהליכים	הורג (סוגר) תהליכים
last		מפסיק (שובר) לולאה
lc	מחרוזות	הופך מחרוזות לאותיות-קטנות (אנגליות)
lcfirst	מחרוזות	מקטין רק את התו הראשון של המחרוזת
length	מחרוזות	מחזיר אורך של מחרוזת
link	קבצים	למערכות יוניקס - יוצר שם חדש לאותו קובץ ישן
listen	רשת	אומר למערכת מה גודל התור המקסימאלי לשרת שלך
local		מגדיר משתנה גלובלי זמני. אם המשתנה הגלובלי כבר קיים, ערכו נשמר בצד ומוגדר משתנה גלובלי זמני חדש. בסוף הבלוק הערך של המשתנה הגלובלי הקודם יוחזר

שם פונקציה	קטגוריה	תיאור
localtime		מתרגם את מה שהוצא ע"י time, לרשימה המכילה את הזמן המקומי באזורך
log	מתמטיקה	מחזיר את הלוגריתם (על בסיס e) למספר
lstat		אם המערכת שלך תומכת ב-symbolic-link, הפונקציה יודעת לעשות stat ללינק
map	מערכים	מעבירה רשימה דרך ביטוי. הביטוי מחזיר עבור כל ערך ברשימה רשימה קטנה של אפס או יותר איברים
mkdir	קבצים	יוצר ספרייה
msgctl	תהליכים	תקשורת בין תהליכים
msgget	תהליכים	תקשורת בין תהליכים
msgrcv	תהליכים	תקשורת בין תהליכים
msgsnd	תהליכים	תקשורת בין תהליכים
my	משתנים	יוצר משתנה מקומי
new	אובייקטים	יוצר אובייקטים, בערך.
next		מקפיץ את התוכנית ישר לסיבוב הבא של לולאה
no		ההפך של use, בערך
oct	מתמטיקה	מקבל מחרוזת המכילה מספר בייצוג אוקטלי, ומחזיר ערך
open	קבצים	פתיחת קובץ
opendir	קבצים	פתיחת ספרייה
ord	מתמטיקה	מחזיר את הערך המספרי (אסקי) של תו
pack		"אריזת" רשימת משתנים לתוך סקלר. שימושי בתקשורת מול תוכניות C או מעל רשת, כאשר הפרוטוקול דורש מבנה מסויים. ראה גם unpack, העושה פעולה הפוכה.
package	מודולים	הצהרת תחילת מודול.
pipe		פותח un-named pipe בין התהליך לבין עצמו. בד"כ אפשר למצוא פקודה זאת רגע לפני fork, כשהתהליך מתפצל ורוצה לשמור על תקשורת בינו לבין הבנים שלו
pop	מערכים	מוציא את האיבר האחרון ממערך. כמו מחסנית. ראה גם unpop, וכן push.
pos	מחרוזות	מחזיר את המיקום בו פקודת החיפוש במחרוזת, m//g עצרה.
print	קלט/פלט	להדפיס פלט למסך/קובץ
printf	קלט/פלט	בול אותו הדבר, רק בסגנון שפת C
push	מערכים	מתייחס למערך כמו אל מחסנית, ודוחף לסופה איברים. ראה גם pop

שם פונקציה	קטגוריה	תיאור
quotemeta	מחרוזות	מחזיר את המחרוזת שהוא קיבל כקלט, כך שלפני כל תו שהוא אינו אלפאנומרי, יש סלש.
rand	מתמטיקה	מחזיר ערך רנדומלי בין 0 ל-1, או למספר נתון
read	קבצים	קריאת מידע מקובץ
readdir	קבצים	קריאת תוכן של ספרייה. ראה גם opendir
readlink	קבצים	אם יש לך symbolic link, הפונקציה תגיד לך לאן הוא מצביע
recv	רשת	מקבל מידע שנשלח ברשת אליך
redo		מחזיר את הלולאה לתחילתה, מבלי לבדוק את התנאי של הלולאה.
ref	משתנים	אם משתנה הוא לא מצביע, מחזיר undef. אם הוא כן, אז אומר לאיזה טיפוס הוא מצביע.
rename	קבצים	שינוי שם לקובץ
require	מודולים	קודם כל, אתה יכול לבדוק בעזרת הוראה זאת את הגרסא של Perl, ולסרב להמשיך אם היא לא תואמת. דבר שני זה כמו לעשות use למודול בלי לעשות import.
reset		פונקציה שאינה כבר בשימוש
return		יציאה מפונקציה או מביטוי eval
reverse	מערכים	מקבל רשימה, ומחזיר את אותה הרשימה, רק בסדר הפוך
rewinddir	קבצים	מחזיר את נקודת הקריאה של ספרייה פתוחה להתחלה (אתה מתחיל בספריות מההתחלה, ואז קורא באופן סדרתי קדימה. צריך את הפונקציה כדי לחזור להתחלה, במקום לסגור ולפתוח מחדש את הספרייה)
rindex	מחרוזות	מחפש תת מחרוזות בתוך מחרוזת, מהסוף להתחלה
rmdir	קבצים	מוחק ספרייה
s///	מחרוזות	אופרטור החלפה במחרוזות. ראה פרק 4
scalar	משתנים	כופה הקשר סקלרי על הביטוי שאחריו. לא קיים ביטוי מקביל שכופה הקשר רשימה
seek	קבצים	שינוי נקודת הקריאה בקובץ. אם מדובר על קובץ טקסט, עם אורך שורה משתנה, מומלץ להשתמש בפונקציה tell, ששומרת את המיקום הנוכחי בתוך קובץ (אם כי תמיד אפשר לקפוץ להתחלת או לסוף הקובץ)
seekdir	קבצים	מכוון את מיקום הקריאה בתוך ספרייה
select	קבצים	ישנה דרך להשפיע על ההתנהגות של קבצים בעזרת משתנים מיוחדים. פקודת select אומרת לאיזה קובץ אתה מתכוון
select	רשת	דרך לעבוד על מספר קישורי רשת, א-סינכרונית
semctl	תהליכים	תקשורת בין תהליכים -דיגלונים
semget	תהליכים	תקשורת בין תהליכים -דיגלונים

שם פונקציה	קטגוריה	תיאור
semop	תהליכים	תקשורת בין תהליכים -דיגלונים
send	רשת	שליחת הודעה על קישור רשת
setpgrp		פונקציה חסרת תועלת
setpriority		שינוי העדיפות של התהליך
setsockopt	רשת	שינוי בפרמטרים של קישור רשת
shift	מערכים	מוציא ממערך את האיבר הראשון. ראה גם unshift
shmctl	תהליכים	תקשורת בין תהליכים - זיכרון משותף
shmget	תהליכים	תקשורת בין תהליכים - זיכרון משותף
shmread	תהליכים	תקשורת בין תהליכים - זיכרון משותף
shmwrite	תהליכים	תקשורת בין תהליכים - זיכרון משותף
shutdown	רשת	סגירת קישור רשת
sin	מתמטיקה	סינוס. בחיך.
sleep		מעכב את התוכנית למשך מספר שניות
socket	רשת	יוצר קישור רשת
socketpair	רשת	יוצר זוג תואם של קישורי רשת. כמו pipe, רק דו כיווני
sort	מערכים	ממין מערך. אפשר גם לקבוע לו מה נחשב "גדול" ומה "קטן"
splice	מערכים	מניפולציות על מערכים, כולל חיתוך והוספה. (פרק 2)
split	מערכים	חיתוך מחרוזת לרשימת מחרוזות
sprintf	מחרוזות	כמו printf, רק במקום להדפיס את התוצאה היא מחזירה אותה
sqrt	מתמטיקה	שורש ריבועי
srand	מתמטיקה	מאפס את רצף המספרים האקראיים.
stat	קבצים	מחזיר מאפיינים של קובץ
study	מחרוזות	אם אתה הולך לטחון מחרוזת מסוימת בהרבה חיפושים, רק מחרוזת יחידה, אז פונקציה זו תאיץ את החיפושים
sub		הצהרה על פונקציה
substr	מחרוזות	מחזירה חתיכה ממחרוזת
symlink		יוצר symbolic link חדש
syscall		מפעיל תוכנית חיצונית
sysopen	קבצים	פתיחת קובץ, בסגנון C
sysread	קבצים	קריאה מקובץ, בסגנון C
system		מפעיל תוכנית חיצונית, ומחזיר את ערך היציאה שלה
syswrite	קבצים	כתיבה לקובץ, בסגנון C

שם פונקציה	קטגוריה	תיאור
tell	קבצים	מחזירה את המיקום שנוכחי של נקודת הקריאה בקובץ שימושי במיוחד אם אתה רוצה לחזור למיקום כלשהו בקובץ טקסט
telldir	קבצים	אותו הדבר, רק לספריות
tie	משתנים	אפשרות מעניינת להתייחס לדברים כאילו הם משתנים למשל, להתייחס למסד נתונים כאילו הוא טבלת האש
tied	משתנים	אם משתנה הוא מקושר, (בעזרת פקודת tie) אזי הפונקציה הזאת תחזיר מצביע לאובייקט אליו המשתנה מקושר
time		מחזיר את מספר השניות שעברו מאז הראשון בינואר, 1970
times		מחזיר את הזמן שהמשתמש והמערכת ניצלו את המעבד
tr///	מחרוזות	אופרטור תרגום. ראה פרק מחרוזות
truncate	קבצים	חותך קובץ פתוח לאורך ספציפי
uc	מחרוזות	מחזיר את המחרוזות שהוא קיבל, כאשר כל האותיות באנגלית נהפכו לאותיות-גדולות
ucfirst	מחרוזות	אותו הדבר, רק מגדיל רק את האות הראשונה במחרוזת
umask		משנה את ההרשאות שיהיו על קבצים חדשים שיווצרו ע"י התוכנית
undef	משתנים	מחזיר ערך בלתי-מוגדר, וגם הופך כל משתנה שהוא מקבל כפרמטר לבעל ערך בלתי מוגדר
unlink	קבצים	מוחק קובץ אחד או יותר
unpack		ההפך של pack. לוקח מחרוזת המכילה מבנה נתונים דחוס, ופורסת אותו למשתנים שונים
unshift	מערכים	דוחף ערכים לתחילת מערך. ההפך של shift
untie	משתנים	מפרק את הקשר בין משתנה לאובייקט. ראה גם tie
use	מודולים	מקשר מודול לתוכנית, או נותן הוראת קומפילר. ראה פרק רלוונטי
utime		משנה את חותמת הזמן של קובץ (מתי נוצר, מתי שונה)
values	טבלאות האש	מחזיר את כל הערכים שבתוך טבלת האש
vec	משתנים	דרך פונקציה זאת, אתה יכול להתייחס למחרוזת כמערך מספרים שלמים בלי סימן. (unsigned integer)
wait		מחכה לתהליך-בן כלשהו לסיים פעולתו
waitpid		מחכה לתהליך-בן ספציפי לסיים פעולתו
wantarray		מחזיר האם פונקציה זו נקראה בהקשר רשימה או סקלר
warn		מדפיס הודעת שגיאה כמו die, אבל לא מפסיק את ריצת התוכנית
write	קלט/פלט	הדפסת פלט בפורמט מיוחד
y///	מחרוזות	בדיוק כמו tr///. ראה פרק רלוונטי. (מחרוזות)